This Maple file accompanies the paper "Compacted binary trees admit a stretched exponential" by Andrew Elvey Price, Wenjie Fang and Michael Wallner.

##

Abstract:
A compacted binary tree is a directed acyclic graph encoding a binary tree in which common subtrees are factored and shared,
such that they are represented only once. We show that the number of compacted binary trees of size n is asymptotically given by

Theta(n! 4^n e^(3 a_1 n^(1/3)) n^(3/4)),

where a_1 is the largest root of the Airy function and approximately equal to 2.338. Our method involves a new two parameter
recurrence which yields an algorithm of quadratic arithmetic complexity. We use empirical methods to estimate the values of all
terms defined by the recurrence, then we prove by induction that these estimates are sufficiently accurate for large n to determine
the asymptotic form. Our results also lead to new bounds on the number of minimal finite automata recognizing a finite language
on a binary alphabet showing the appearance of a stretched exponential.

##

In this worksheet
*) we compute the initial terms of the recurrences
*) we compute the expansions of P_{n,m} and Q_{n,m} and create Figures 9-12
*) we verify Lemmas 4.5 and 5.4

```
> restart;
  with(plots):
```

Recurrences

In some computations at the end we use the gfun-package.
It is just needed to guess simple functional equations for generating functions and manipulate their coefficient.
This is not needed in the remainder of this worksheet.

For the latest version see Salvy's homepage http://perso.ens-lyon.fr/bruno.salvy/software/the-gfun-package/.
For these computations gfun verison 3.76 was used.

```
> #libname := "<path>\\gfun.mla", libname;    #set the path to gfun.mla
  libname := "D:\\lib\\gfun.mla", "D:\\lib", "/home/michaelwallner/lib/gfun.mla", "home/michaelwallner/lib", libname;
  with(gfun): gfun:-version;
```

libname := "D:\lib\gfun.mla", "D:\lib", "/home/michaelwallner/lib/gfun.mla", "home/michaelwallner/lib", "C:\Program Files\Maple 18\lib", "."

3.76 **(1.1)**

Maximal number of computed terms

```
> NN := 100;
```
$NN := 100$ (1.2)

Relaxed binary trees (Section 2.1)

Compute the relaxed binary trees of size up to NN

```
> for n from 0 to NN do
  for m from 0 to NN do
     rr[n,m] := 0:
  end:
  end:

  for n from 0 to NN do
     rr[n,0] := 1:
  end:

  for n from 1 to NN do
     for m from 1 to n do
        rr[n,m] := rr[n,m-1]+(m+1)*rr[n-1,m];
     end do:
  end do:
```

print the array

```
> for m from 7 to 0 by -1 do
  for n from 0 to 10 do
  printf("%10.0f ",rr[n,m]);
  end;
  printf("\n");
  end;
```

```
         0          0          0          0          0
         0          0     311250    6173791   86626584
1035808538
         0          0          0          0          0
         0      18628     311250    3683791   37236256
342795866
         0          0          0          0          0
      1363      18628     180854    1505041   11449719
82142074
         0          0          0          0        127
      1363      10450      69086     419917    2419473
13443760
         0          0          0         16        127
       728       3635      16836      74487     319888
1346395
         0          0          3         16         63
       220        723       2296       7143      21940
66843
         0          1          3          7         15
        31         63        127        255        511
1023
         1          1          1          1          1
         1          1          1          1          1
 1
```

check

```
> seq(rr[n,n],n=0..min(NN,12));
```

1, 1, 3, 16, 127, 1363, 18628, 311250, 6173791, 142190703, 3737431895, 110577492346, 3641313700916 (1.1.1)

these are the correct numbers,

see http://oeis.org/A082161

```
> 1, 3, 16, 127, 1363, 18628, 311250, 6173791, 142190703,
  3737431895, 110577492346, 3641313700916;
```

1, 3, 16, 127, 1363, 18628, 311250, 6173791, 142190703, 3737431895, 110577492346, 3641313700916 **(1.1.2)**

Compacted binary trees (Section 2.2)

Compute the relaxed binary trees of size up to NN

```
> for n from 0 to NN do
  for m from 0 to NN do
     cc[n,m] := 0:
  end:
  end:

  for n from 0 to NN do
     cc[n,0] := 1:
  end:

  for n from 1 to NN do
     for m from 1 to n do
        cc[n,m] := cc[n,m-1]+(m+1)*cc[n-1,m]-(m-1)*cc[n-2,
  m-1];
     end do:
  end do:
```

print the array

```
> for m from 7 to 0 by -1 do
  for n from 0 to 10 do
  printf("%10.0f ",cc[n,m]);
  end;
  printf("\n");
  end;
```

```
         0          0          0          0          0
     0          0     230943    4395855   59662737
694358115
         0          0          0          0          0
     0      14487     230943    2635233   25881555
232867617
         0          0          0          0          0
  1119      14487     135129    1091067    8110569
57152067
         0          0          0          0        111
  1119       8217      52683     313161    1774899
9741297
         0          0          0         15        111
   609       2955      13425      58611     249369
1042635
         0          0          3         15         57
   195        633       1995       6177      18915
57513
         0          1          3          7         15
    31         63        127        255        511
1023
         1          1          1          1          1
     1          1          1          1          1
 1
```

check

```
> seq(cc[n,n],n=0..min(NN,12));
```

1, 1, 3, 15, 111, 1119, 14487, 230943, 4395855, 97608831, 2482988079, 71321533887, 2286179073663 **(1.2.1)**

these are the correct numbers,
see http://oeis.org/A254789

```
> 1, 1, 3, 15, 111, 1119, 14487, 230943, 4395855, 97608831,
  2482988079, 71321533887, 2286179073663;
```

1, 1, 3, 15, 111, 1119, 14487, 230943, 4395855, 97608831, 2482988079, 71321533887, 2286179073663 **(1.2.2)**

Weighted Dyck meanders (Section 3.2)

Maximal number of meanders computed

```
> NN:=100:
```

compute the weighted Dyck meanders of size up to NN

```
> for n from 0 to NN do
  for m from -1 to NN do
  dd[n,m] := 0:
  end:end:

  dd[0,0] := 1:
  for n from 1 to NN do
     for m from 0 to n do
        dd[n,m] := (n-m+2)/(n+m)*dd[n-1,m-1]+dd[n-1,m+1];
     end do:
  end do:
```

print the array

```
> for m from 5 to 0 by -1 do
  for n from 0 to 10 do
  printf("%10.0f ",dd[n,m]*factorial(floor((n+m)/2)));
  end;
  printf("\n");
  end;
         0          0          0          0          0
     1          0          63          0          2296
 0
         0          0          0          0          1
     0          31          0          723          0
16836
         0          0          0          1          0
    15          0          220          0          3635
 0
         0          0          1          0          7
     0          63          0          728          0
10450
         0          1          0          3          0
    16          0          127          0          1363
 0
         1          0          1          0          3
     0          16          0          127          0
1363
```

check

```
> seq(factorial(n/2)*dd[n,0],n=0..min(NN,22),2);
  seq(rr[n,n],n=0..min(NN,11));
```

1, 1, 3, 16, 127, 1363, 18628, 311250, 6173791, 142190703, 3737431895, 110577492346

1, 1, 3, 16, 127, 1363, 18628, 311250, 6173791, 142190703, 3737431895, 110577492346 **(1.3.1)**

explicit formulas

```
> n:='n':'r(n,n)' = 1/factorial(n);
  seq(dd[n,n]-1/factorial(n),n=0..10);
```

$$r(n, n) = \frac{1}{n!}$$

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 **(1.3.2)**

```
> n:='n':'r(n,n-2)' = (2^(n-1)-1)/factorial(n-1);
  seq(dd[n,n-2]-(2^(n-1)-1)/factorial(n-1),n=2..10);
```

$$r(n, n-2) = \frac{2^{n-1}-1}{(n-1)!}$$

0, 0, 0, 0, 0, 0, 0, 0, 0 **(1.3.3)**

```
> seq(dd[n,n-4]*factorial(n-2),n=4..20);
  listtorec([%],a(n));
  rsolve(op(1,%),a(n));
  subs(n=n-4,%);
```

3, 16, 63, 220, 723, 2296, 7143, 21940, 66843, 202576, 611823, 1843660, 5547363, 16674856, 50090103, 150401380, 451466283

$[\{-6\, a(n) + 11\, a(n+1) - 6\, a(n+2) + a(n+3), a(0) = 3, a(1) = 16, a(2) = 63\}, \mathit{ogf}]$

$$\frac{1}{2} + \frac{21}{2}\, 3^{n} - 8\, 2^{n}$$

$$\frac{1}{2} + \frac{21}{2}\, 3^{n-4} - 8\, 2^{n-4}$$ **(1.3.4)**

```
> n:='n':'r(n,n-4)' = (21*3^(n-4)+1-2^n)/2/factorial(n-2);
  seq(dd[n,n-4]-rhs(%),n=4..10);
```

$$r(n, n-4) = \frac{1}{2}\, \frac{21\, 3^{n-4} + 1 - 2^{n}}{(n-2)!}$$

0, 0, 0, 0, 0, 0, 0 **(1.3.5)**

```
> seq(dd[n,n-6]*factorial(n-3),n=6..20);
  listtorec([%],a(n));
  rsolve(op(1,%),a(n));
  subs(n=n-6,%);
```

16, 127, 728, 3635, 16836, 74487, 319888, 1346395, 5588156, 22964447, 93701448, 380353155, 1538087476, 6202440007, 24960161408

$[\{24\, a(n) - 50\, a(n+1) + 35\, a(n+2) - 10\, a(n+3) + a(n+4), a(0) = 16, a(1) = 127, a(2) = 728, a(3) = 3635\}, \mathit{ogf}]$

$$16\, 2^{n} - \frac{189}{2}\, 3^{n} - \frac{1}{6} + \frac{284}{3}\, 4^{n}$$

$$16\, 2^{n-6} - \frac{189}{2}\, 3^{n-6} - \frac{1}{6} + \frac{284}{3}\, 4^{n-6}$$ **(1.3.6)**

```
> n:='n':'r(n,n-6)' = factor((-189*3^(n-6)*(1/2)-1/6+284*4^
  (n-6)*(1/3)+16*2^(n-6))/factorial(n-3));
  seq(dd[n,n-6]-rhs(%),n=6..10);
```

$$r(n, n-6) = \frac{1}{6}\, \frac{96\, 2^{n-6} - 567\, 3^{n-6} - 1 + 568\, 4^{n-6}}{(n-3)!}$$

0, 0, 0, 0, 0 **(1.3.7)**

Weighted compacted Dyck meanders (Section 5)

Maximal number of meanders computed

```
> NN:=100:
```

compute the meanders of size up to NN

```
> for n from -1 to NN do
  for m from -1 to NN do
     ee[n,m] := 0:
  end:
  end:

  ee[0,0] := 1:
  ee[1,1] := 1:
  for n from 2 to NN do
     ee[n,0] := ee[n-1,1];
     for m from 1 to n do
        ee[n,m] := (n-m+2)/(n+m)*ee[n-1,m-1]+ee[n-1,m+1]-2*
  (n-m-2)/(n+m)/(n+m-2)*ee[n-3,m-1];
     end:
  end:
```

print the array

```
> for m from 5 to 0 by -1 do
  for n from 0 to 9 do
  printf("%10.0f ",ee[n,m]*factorial(floor((n+m)/2)));
  end;
  printf("\n");
  end;
          0          0          0          0          0
     1          0         63          0       1995
          0          0          0          0          1
     0         31          0        633          0
          0          0          0          1          0
    15          0        195          0       2955
          0          0          1          0          7
     0         57          0        609          0
          0          1          0          3          0
    15          0        111          0       1119
          1          0          1          0          3
     0         15          0        111          0
```

check

```
> seq(factorial(n/2)*ee[n,0],n=0..min(NN,22),2);
  seq(cc[n,n],n=0..min(NN,11));
```

1, 1, 3, 15, 111, 1119, 14487, 230943, 4395855, 97608831, 2482988079, 71321533887

1, 1, 3, 15, 111, 1119, 14487, 230943, 4395855, 97608831, 2482988079, 71321533887 **(1.4.1)**

We guess a few simple formulas for the numbers ending close to the upper end.

There is just 1 path ending on the very top

```
> seq(ee[n,n]*factorial(n),n=2..20);
```

1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1, 1 **(1.4.2)**

```
> n:='n':
  seq(ee[n,n-2]*factorial(n-1),n=2..20);
  listtorec([%],a(n));
```

```
rsolve(op(1,%),a(n));
subs(n=n-2,%);
```

1, 3, 7, 15, 31, 63, 127, 255, 511, 1023, 2047, 4095, 8191, 16383, 32767, 65535, 131071, 262143, 524287

$$[\{2\,a(n) - 3\,a(n+1) + a(n+2), a(0) = 1, a(1) = 3\}, ogf]$$

$$2\,2^n - 1$$

$$2\,2^{n-2} - 1 \quad (1.4.3)$$

```
> n:='n':
seq(ee[n,n-4]*factorial(n-2),n=4..20);
listtorec([%],a(n));
rsolve(op(1,%),a(n));
subs(n=n-4,%);
```

3, 15, 57, 195, 633, 1995, 6177, 18915, 57513, 174075, 525297, 1582035, 4758393, 14299755, 42948417, 128943555, 387027273

$$[\{6\,a(n) - 5\,a(n+1) + a(n+2), a(0) = 3, a(1) = 15\}, ogf]$$

$$9\,3^n - 6\,2^n$$

$$9\,3^{n-4} - 6\,2^{n-4} \quad (1.4.4)$$

```
> n:='n':
seq(ee[n,n-6]*factorial(n-3),n=6..20);
listtorec([%],a(n));
rsolve(op(1,%),a(n));
subs(n=n-6,%);
```

15, 111, 609, 2955, 13425, 58611, 249369, 1042635, 4306785, 17637411, 71783529, 290841915, 1174503345, 4731445011, 19026124089

$$[\{-24\,a(n) + 26\,a(n+1) - 9\,a(n+2) + a(n+3), a(0) = 15, a(1) = 111, a(2) = 609\}, ogf]$$

$$72\,4^n + 6\,2^n - 63\,3^n$$

$$72\,4^{n-6} + 6\,2^{n-6} - 63\,3^{n-6} \quad (1.4.5)$$

ehat from Section 5.2 and the proof of the second upper bound of Lemma 5.1

```
> for n from -1 to NN do
  for m from -1 to NN do
     eh[n,m] := 0:
  end:
  end:

  for n from -1 to NN do
     eh[n,0] := ee[n,0];
     eh[n,1] := ee[n,1];
     eh[n,2] := ee[n,2];
     eh[n,n] := ee[n,n];
  end:

  for n from 3 to NN do
     eh[n,0] := eh[n-1,1];
     for m from 1 to n-1 do
        eh[n,m] := (n-m+2)/(n+m)*eh[n-1,m-1]+(n-m-2)/(n-m)
  *eh[n-1,m+1]+2/(n-m)*eh[n-2,m+2]+2/(n+m)*eh[n-3,m+1]+4/
  (n+m)/(n+m-2)*eh[n-3,m-1];
     end:
```

```
end:
> for m from 5 to 0 by -1 do
for n from 0 to 9 do
printf("%10.2f ",(eh[n,m]-ee[n,m])*factorial(floor((n+m)/2)));
end;
printf("\n");
end;
```

```
      0.00       0.00       0.00       0.00       0.00
    0.00       0.00       0.00       0.00       0.00
      0.00       0.00       0.00       0.00       0.00
    0.00       0.00       0.00       0.00       0.00
      0.00       0.00       0.00       0.00       0.00
    0.00       0.00       0.00       0.00      65.33
      0.00       0.00       0.00       0.00       0.00
    0.00       0.00       0.00      13.33       0.00
      0.00       0.00       0.00       0.00       0.00
    0.00       0.00       2.00       0.00      33.50
      0.00       0.00       0.00       0.00       0.00
    0.00       0.00       0.00       2.00       0.00
```

This should always be negative

```
> for n from -1 to NN do
for m from -1 to NN do
    ehdiff[n,m] := 0:
end:
end:

for n from 3 to NN do
    for m from 1 to n-1 do
        ehdiff[n,m] := (-2)/(n-m)*eh[n-1,m+1]+2/(n-m)*eh[n-2,m+2]+2/(n+m)*eh[n-3,m+1]+4/(n+m)/(n+m-2)*eh[n-3,m-1];
    end:
end:
> mysign := proc(i) if i=0 then return 0 else return sign(i) end end:
```

looks good

```
> for m from 16 to 0 by -1 do
for n from 0 to 20 do
printf("%3.0f ",mysign(ehdiff[n,m]*factorial(floor((n+m)/2))));
end;
printf("\n");
end;
```

```
  0   0   0   0   0   0   0   0   0   0   0   0   0   0
  0   0   0   0   0   0  -1
  0   0   0   0   0   0   0   0   0   0   0   0   0   0
  0   0   0   0   0  -1   0
  0   0   0   0   0   0   0   0   0   0   0   0   0   0
  0   0   0   0  -1   0  -1
  0   0   0   0   0   0   0   0   0   0   0   0   0   0
  0   0   0  -1   0  -1   0
  0   0   0   0   0   0   0   0   0   0   0   0   0   0
  0   0  -1   0  -1   0  -1
  0   0   0   0   0   0   0   0   0   0   0   0   0   0
  0  -1   0  -1   0  -1   0
  0   0   0   0   0   0   0   0   0   0   0   0   0   0
 -1   0  -1   0  -1   0  -1
```

```
  0   0   0   0   0   0   0   0   0   0   0   0   0  -1
  0  -1   0  -1   0  -1   0
  0   0   0   0   0   0   0   0   0   0   0   0  -1   0
 -1   0  -1   0  -1   0  -1
  0   0   0   0   0   0   0   0   0   0   0  -1   0  -1
  0  -1   0  -1   0  -1   0
  0   0   0   0   0   0   0   0   0   0  -1   0  -1   0
 -1   0  -1   0  -1   0  -1
  0   0   0   0   0   0   0   0   0  -1   0  -1   0  -1
  0  -1   0  -1   0  -1   0
  0   0   0   0   0   0   0   0  -1   0  -1   0  -1   0
 -1   0  -1   0  -1   0  -1
  0   0   0   0   0   0   0  -1   0  -1   0  -1   0  -1
  0  -1   0  -1   0  -1   0
  0   0   0   0   0   0  -1   0  -1   0  -1   0  -1   0
 -1   0  -1   0  -1   0  -1
  0   0   0   0   0  -1   0  -1   0  -1   0  -1   0  -1
  0  -1   0  -1   0  -1   0
  0   0   0   0   0   0   0   0   0   0   0   0   0   0
  0   0   0   0   0   0   0
```

```
> n0:=51:
  #seq(ehdiff[n0,m],m=n0 mod 2..n0,2);
  seq(mysign(ehdiff[n0,m]),m=n0 mod 2..n0,2);
```

-1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, -1, 0, 0 **(1.4.1.1)**

```
> m := 'm':
  n := 'n':
```

This should also be negative

```
> for n from -1 to NN do
  for m from -1 to NN do
     eediff[n,m] := 0:
  end:
  end:

  for n from 3 to NN do
     for m from 1 to n-3 do
        eediff[n,m] := (-2)/(n-m)*ee[n-1,m+1]+2/(n-m)*ee
  [n-2,m+2]+2/(n+m)*ee[n-3,m+1]+4/(n+m)/(n+m-2)*ee[n-3,
  m-1];
     end:
  end:
> for m from 16 to 0 by -1 do
  for n from 0 to 20 do
  printf("%3.0f ",mysign(eediff[n,m]));
  end;
  printf("\n");
  end;
```

```
  0   0   0   0   0   0   0   0   0   0   0   0   0   0
  0   0   0   0   0   0  -1
  0   0   0   0   0   0   0   0   0   0   0   0   0   0
  0   0   0   0   0  -1   0
  0   0   0   0   0   0   0   0   0   0   0   0   0   0
  0   0   0   0  -1   0  -1
  0   0   0   0   0   0   0   0   0   0   0   0   0   0
  0   0   0  -1   0  -1   0
  0   0   0   0   0   0   0   0   0   0   0   0   0   0
```

```
 0   0  -1   0  -1   0  -1
 0   0   0   0   0   0   0   0   0   0   0   0   0   0
 0  -1   0  -1   0  -1   0
 0   0   0   0   0   0   0   0   0   0   0   0   0   0
-1   0  -1   0  -1   0  -1
 0   0   0   0   0   0   0   0   0   0   0   0   0  -1
 0  -1   0  -1   0  -1   0
 0   0   0   0   0   0   0   0   0   0   0   0  -1   0
-1   0  -1   0  -1   0  -1
 0   0   0   0   0   0   0   0   0   0   0  -1   0  -1
 0  -1   0  -1   0  -1   0
 0   0   0   0   0   0   0   0   0   0  -1   0  -1   0
-1   0  -1   0  -1   0  -1
 0   0   0   0   0   0   0   0   0  -1   0  -1   0  -1
 0  -1   0  -1   0  -1   0
 0   0   0   0   0   0   0   0  -1   0  -1   0  -1   0
-1   0  -1   0  -1   0  -1
 0   0   0   0   0   0   0  -1   0  -1   0  -1   0  -1
 0  -1   0  -1   0  -1   0
 0   0   0   0   0   0  -1   0  -1   0  -1   0  -1   0
-1   0  -1   0  -1   0  -1
 0   0   0   0   0  -1   0  -1   0  -1   0  -1   0  -1
 0  -1   0  -1   0  -1   0
 0   0   0   0   0   0   0   0   0   0   0   0   0   0
 0   0   0   0   0   0   0
```

Proof for ee by induction

The proper contributions from the first 2 terms of the recurrence for e directly give the recurrence for d;
it remains to show that the following contribution is non-positive (note that we already used the inductive hypothesis that e(n,m) <= d(n,m)

```
> NegPart := -2/(n-m)*d(n-1,m+1) + 2/(n-m)*d(n-2,m+2) + 2/(n+m)*d(n-3,m+1) + 4/(n+m)/(n+m-2)*d(n-3,m-1);
```

$$NegPart := -\frac{2\,d(n-1,m+1)}{n-m} + \frac{2\,d(n-2,m+2)}{n-m} + \frac{2\,d(n-3,m+1)}{n+m} + \frac{4\,d(n-3,m-1)}{(n+m)\,(n+m-2)} \qquad \textbf{(1.4.1.2)}$$

This is the recurrence for d

```
> ddrec := d(n,m) = (n-m+2)/(n+m)*d(n-1,m-1) + d(n-1,m+1);
```

$$ddrec := d(n,m) = \frac{(n-m+2)\,d(n-1,m-1)}{n+m} + d(n-1,m+1) \qquad \textbf{(1.4.1.3)}$$

This is indeed negative

```
> subs(n=n-1,m=m+1,ddrec);
  subs(n=n-2,m=m,ddrec);
  NNee1 := collect(subs(%%,%,NegPart),e,factor);
```

$$d(n-1,m+1) = \frac{(n-m)\,d(n-2,m)}{n+m} + d(n-2,m+2)$$

$$d(n-2,m) = \frac{(n-m)\,d(n-3,m-1)}{n+m-2} + d(n-3,m+1)$$

$$NNee1 := \frac{2\,(-n+m+2)\,d(n-3,m-1)}{(n+m)\,(n+m-2)} \qquad \textbf{(1.4.1.4)}$$

```
> m := 'm':
  n := 'n':
```

Newton polygons

Programs

look at possible degrees of each term in n and m
x-axis m degree
y-axis n degree

```
> mynewt := proc(F,m,n)
     local ss,newt,tt,mdeg,ndeg;
     ss := 0;
     newt := {}:
     for tt in expand(collect(F,[m,n],simplify)) do
        mdeg := simplify(m*diff(tt,m)/tt);
        ndeg := simplify(n*diff(tt,n)/tt);
        newt := {op(newt), [mdeg,ndeg]};
     end:
     return newt;
  end:
```

Get the term of order Theta(m^a*n^b) from the Newton polygon computed using mynewt

```
> getel := proc(tmp,a,b)
     local tt,mdeg,ndeg,ret:
     ret := 0:
     for tt in expand(tmp) do
        mdeg := m*diff(tt,m)/tt;
        ndeg := n*diff(tt,n)/tt;
        if mdeg=a and ndeg=b then ret := ret + tt end;
     end:
     return ret;
  end:
```

Get the maximal power of n^b for each m^a for a=0..M

```
> getMaxNewt := proc(M::posint, newt)
     local i, el, mnmax;

     for i from 0 to M do mnmax[i]:=-infinity end:
     for el in newt do
        if el[1]<=M then
        if mnmax[el[1]]<el[2] then mnmax[el[1]]:=el[2] end:
     end;
     end;

     return mnmax;
  end:
```

Compute the slopes and corners of the convex hull

```
> maxslope := proc(ll,M,i)
     local j,sl,tmp,sj;
     sl := ll[i+1]-ll[i]:   #initial slope between first 2
  points
     #is this the one of the hull? find max starting from 0
     sj:=M;
     for j from i+1 to M do
        tmp := (ll[j]-ll[i])/(j-i);
        if tmp >= sl then sl:=tmp;sj:=j; end;
     end:
     return sl,sj;
  end:
```

Compute the slopes and corners of the convex hull

```
> getslopes := proc(ll,M)
```

```
    local sl,sj,li,ls;
    sj:=0;
    li:=sj:
    ls:=0:
    #go on and find other slopes of convex hull
    while sj<M do
       sl,sj := maxslope(ll,M,sj):
       #save it
       li:=li,sj;
       ls:=ls,sl;
    end:

    li:=[li];
    ls:=[seq(ls[i],i=2..nops(li))];
    return ls,li;
 end:
```

Shorthands

The largest root of the Airy funciton Ai(z)

```
> A1 := AiryAiZeros(1);
  evalf(%);
```

$$A1 := \mathrm{AiryAiZeros}(1)$$
$$-2.338107410 \qquad (2.2.1)$$

This is the constant c of Equation (6):
s(n) = 2 + c*n^(-2/3) + ...
(Also called sigma_2 in the ansatz used in the proof of Lemma 4.2)

```
> csubs := isolate((1/2)*2^(1/3)*(c) = a1,c);
```

$$csubs := c = a1\, 2^{2/3} \qquad (2.2.2)$$

Labeling options to produce nice plots.

```
> myoptionsLo := labels=["i", "j"],symbolsize=25, symbol=
  diamond,axesfont = ["HELVETICA", "ROMAN", 15],labelfont =
  ["HELVETICA", 18]:
  myoptionsUp := labels=["i", "j"],symbolsize=20, symbol=
  diamond,axesfont = ["HELVETICA", "ROMAN", 15],labelfont =
  ["HELVETICA", 18]:
```

We introduce the following shorthands for the Airy function and its derivative

```
> kaplam := AiryAi(AiryAiZeros(1)+2^(1/3)*m/n^(1/3))=kappa,
  AiryAi(1, AiryAiZeros(1)+2^(1/3)*m/n^(1/3))=lambda;
```

$$kaplam := \mathrm{AiryAi}\left(\mathrm{AiryAiZeros}(1) + \frac{2^{1/3}\, m}{n^{1/3}}\right) = \kappa, \mathrm{AiryAi}\left(1, \mathrm{AiryAiZeros}(1) + \frac{2^{1/3}\, m}{n^{1/3}}\right) = \lambda \qquad (2.2.3)$$

Expansions

We expand the Airy function around a1+2^(1/3)*m/n^(1/3) up to chosen order ordAi

```
> FFy := AiryAi(AiryAiZeros(1)+2^(1/3)*m/n^(1/3)+y);
```

$$FFy := \mathrm{AiryAi}\left(\mathrm{AiryAiZeros}(1) + \frac{2^{1/3}\, m}{n^{1/3}} + y\right) \qquad (2.3.1)$$

We use two different expansion orders for the upper and lower bound
(to speed up the computations, and to produce the best pictures)
Here, we start with the **lower bound**, which needs less terms

```
> ordAiLo := 13;
  FFyserLo := map(expand,series(FFy,y,ordAiLo)):
```

$$ordAiLo := 13 \tag{2.3.2}$$

For y = x-(a1+2^(1/3)*m/n^(1/3)) we have that FF(x)=Ai(x),
i.e. an expansion of the Airy function

```
> FFxserLo := subs(y=x-(AiryAiZeros(1)+2^(1/3)*m/n^(1/3)),
  FFyserLo):
```

Replace the appearing Airy functions by our shorthands kappa and lambda

```
> indets(FFxserLo);
  FFxserLoKL := subs(kaplam,AiryAiZeros(1)=a1,FFxserLo):
  indets(%);
```

$$\left\{m, n, x, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{2/3}}, \frac{1}{n^{1/3}}, \mathrm{AiryAi}\left(\mathrm{AiryAiZeros}(1) + \frac{2^{1/3} m}{n^{1/3}}\right), \mathrm{AiryAi}\left(1, \mathrm{AiryAiZeros}(1) + \frac{2^{1/3} m}{n^{1/3}}\right)\right\}$$

$$\left\{a1, \kappa, \lambda, m, n, x, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{2/3}}, \frac{1}{n^{1/3}}\right\} \tag{2.3.3}$$

Then, we use the generic ansatz for factor of FFy, i.e. the Airy function

```
> facAiryLo := (1+(add(q[i]*m^i,i=0..2))/(n));
```

$$facAiryLo := 1 + \frac{q_2 m^2 + q_1 m + q_0}{n} \tag{2.3.4}$$

This is our ansatz
(only the substitution is influenced by the parameters, i.e. not the ms and ns that are already in FFxser;
that is what we want, as all of them should be expanded at a1+2^(1/3)*m/n(1/3))
Note that due to the replacement with kappa and lambda, the expansions are fixed already at this point and the ms and ns in the arguments of Ai and Ai' are not influenced.
Note that we substitute now a1+2^(1/3)*(m+1)/n^(1/3), i.e. m+1 instead of m around which we expanded above.

```
> XFL := (n0,m0) -> subs(n=n0,m=m0,facAiryLo)*subs(x=a1+2^
  (1/3)*(m0+1)/n0^(1/3),FFxserLoKL);
```

$$XFL := (n0, m0) \mapsto subs(n = n0, m = m0, facAiryLo)\ subs\left(x = a1 + \frac{2^{1/3} (m0 + 1)}{n0^{1/3}}, FFxserLoKL\right) \tag{2.3.5}$$

Then, we do the same for the **upper bound**, with a few more terms

```
> ordAiUp := 19;
  FFyserUp := map(expand,series(FFy,y,ordAiUp)):
```

$$ordAiUp := 19 \tag{2.3.6}$$

For y = x-(a1+2^(1/3)*m/n^(1/3)) we have that FF(x)=Ai(x),
i.e. an expansion of the Airy function

```
> FFxserUp := subs(y=x-(AiryAiZeros(1)+2^(1/3)*m/n^(1/3)),
  FFyserUp):
```

Replace the appearing Airy functions by our shorthands kappa and lambda

```
> indets(FFxserUp);
  FFxserUpKL := subs(kaplam,AiryAiZeros(1)=a1,FFxserUp):
  indets(%);
```

$$\left\{m, n, x, \frac{1}{n^{8/3}}, \frac{1}{n^{7/3}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{2/3}}, \frac{1}{n^{1/3}}, \mathrm{AiryAi}\left(\mathrm{AiryAiZeros}(1) + \frac{2^{1/3} m}{n^{1/3}}\right), \mathrm{AiryAi}\left(1, \mathrm{AiryAiZeros}(1) + \frac{2^{1/3} m}{n^{1/3}}\right)\right\}$$

$$\left\{a1, \kappa, \lambda, m, n, x, \frac{1}{n^{8/3}}, \frac{1}{n^{7/3}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{2/3}}, \frac{1}{n^{1/3}}\right\} \quad (2.3.7)$$

In the upper bound, we use a more generic factor of FFy, there are the additional terms p[i] which are divided by n^2

```
> facAiryUp := (1+(add(p[i]*m^i,i=0..4))/(n^2)+(add(q[i]*
  m^i,i=0..2))/(n));
```

$$facAiryUp := 1 + \frac{p_4 m^4 + p_3 m^3 + p_2 m^2 + p_1 m + p_0}{n^2} + \frac{q_2 m^2 + q_1 m + q_0}{n} \quad (2.3.8)$$

This is our ansatz for the upper bound (comparable to XFL)
(only the substitution is influenced by the parameters, i.e. not the ms and ns that are already in FFxser;
that is what we want, as all of them should be expanded at a1+2^(1/3)*m/n(1/3))

```
> XFU := (n0,m0) -> subs(n=n0,m=m0,facAiryUp)*subs(x=a1+2^
  (1/3)*(m0+1)/n0^(1/3),FFxserUpKL);
```

$$XFU := (n0, m0) \mapsto subs(n = n0, m = m0, facAiryUp)\ subs\left(x = a1 + \frac{2^{1/3} (m0 + 1)}{n0^{1/3}}, FFxserUpKL\right) \quad (2.3.9)$$

Finally, the ansatz for the **quotient** of h(n)/h(n-1).
Note that pterm is a mnemonic for "polynomial term", as this value influences the polynomial term n^{alpha};
The other values have similar interpretations:
a .... exponantial growth, i.e. a^n
b .... b will be zero
c .... stretched exponential
d .... technical choice, to simplify the proofs

```
> SF := n -> a+b/n^(1/3)+c/n^(2/3)+pterm/n + d/n^(7/6);
```

$$SF := n \mapsto a + \frac{b}{n^{1/3}} + \frac{c}{n^{2/3}} + \frac{pterm}{n} + \frac{d}{n^{7/6}} \quad (2.3.10)$$

Relaxed Trees - Lower bound (Lemma 4.2)

Xtilde

This is Y tilde

```
> Xansatz := (n,m) -> (1-(4*m^2-3*m+4)/(6*n))*AiryAi(a1+2^
  (1/3)*(m+1)/n^(1/3));
```

$$Xansatz := (n, m) \mapsto \left(1 - \frac{4 m^2 - 3 m + 4}{6 n}\right) \mathrm{AiryAi}\left(a1 + \frac{2^{1/3} (m + 1)}{n^{1/3}}\right) \quad (2.4.1.1)$$

```
> Sansatz := n -> 2 + c*n^(-2/3) + pterm/n - 1/(n^(7/6));
```

(2.4.1.2)

$$Sansatz := n \mapsto 2 + \frac{c}{n^{2/3}} + \frac{pterm}{n} - \frac{1}{n^{7/6}} \quad (2.4.1.2)$$

lower bound

```
> posansatz := -XX(n,m)*SS(n) + (n-m+2)/(n+m)*XX(n-1,m-1)
  + XX(n-1,m+1);
```

$$posansatz := -XX(n, m)\, SS(n) + \frac{(n-m+2)\, XX(n-1, m-1)}{n+m} + XX(n - 1, m+1) \quad (2.4.1.3)$$

```
> posXS := map(simplify,subs(XX=Xansatz,SS=Sansatz,
  posansatz)):
```

For a large n this function of m seems to be positive

```
> Digits:=20:
  e1 := subs(csubs,pterm=8/3,a1=A1,posXS):
  N := 100000;
  M := floor(N^(1/2));
  P1 := plot([seq([mm,(subs(n=N,m=mm,e1))],mm=0..1000)])
  :display(P1);
```

$$N := 100000$$

$$M := 316$$



Prove it

We start with the ansatz of Ytilde in Lemma 5.2.

Recall the general ansatz

```
> facAiryLo*Airy(a1+2^(1/3)*(m+1)/n^(1/3));
  SF(n);
```

$$\left(1+\frac{m^2 q_2+m\,q_1+q_0}{n}\right)\mathit{Airy}\left(a1+\frac{2^{1/3}(m+1)}{n^{1/3}}\right)$$

$$a+\frac{b}{n^{1/3}}+\frac{c}{n^{2/3}}+\frac{pterm}{n}+\frac{d}{n^{7/6}} \qquad \textbf{(2.4.2.1)}$$

Substitute ansatz into the sequence we want to be positive for large n and all m

```
> posF := map(expand, subs(XX=XFL,SS=SF,posansatz)):indets
  (%);
```

$$\left\{a, a1, b, c, d, \kappa, \lambda, m, n, pterm, q_0, q_1, q_2, \frac{1}{n^{7/6}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{2/3}}, \frac{1}{n^{1/3}}, \frac{1}{(n-1)^{1/3}}\right\} \qquad \textbf{(2.4.2.2)}$$

The error terms are (to check, expand posF)

```
> simplify(O((2^(1/3)*(m+1)/n^(1/3)-2^(1/3)*m/n^(1/3))
  ^ordAiLo));
  simplify(O((2^(1/3)*(m)/(n-1)^(1/3)-2^(1/3)*m/n^(1/3))
  ^ordAiLo));
  simplify(O((2^(1/3)*(m+2)/(n-1)^(1/3)-2^(1/3)*m/n^(1/3))
  ^ordAiLo));
```

$$\mathrm{O}\left(\frac{16\,2^{1/3}}{n^{13/3}}\right)$$

$$\mathrm{O}\left(-\frac{16\,2^{1/3} m^{13}\left((n-1)^{1/3}-n^{1/3}\right)^{13}}{(n-1)^{13/3} n^{13/3}}\right)$$

$$\mathrm{O}\left(\frac{16\left(-m\,(n-1)^{1/3}+n^{1/3}(m+2)\right)^{13} 2^{1/3}}{n^{13/3}(n-1)^{13/3}}\right) \qquad \textbf{(2.4.2.3)}$$

remove error terms

```
> posFd := convert(posF,polynom):
```

Next we fit the display size of the plots in the n (Nord) and m (Mord) direction.
(Note that everything up to ordAi is computed, but possibly not shown)

```
> Nord := -ordAiLo/3;
  Mord := floor(ordAiLo/3)+1;
  myview := view=[0..Mord,Nord..0]:
```

$$Nord := -\frac{13}{3}$$

$$Mord := 5 \qquad \textbf{(2.4.2.4)}$$

Expand again with respect to n,
these are then our unknowns

```
> posFe := series(posFd,n=infinity,ceil(-Nord)+1):indets
  (%);
  posFf := convert(%%,polynom):
```

$$\left\{a, a1, b, c, d, \kappa, \lambda, m, n, pterm, q_0, q_1, q_2, \left(\frac{1}{n}\right)^{1/3}, \left(\frac{1}{n}\right)^{2/3}, \left(\frac{1}{n}\right)^{3/2}, \right. \qquad \textbf{(2.4.2.5)}$$

$$\left(\frac{1}{n}\right)^{4/3}, \left(\frac{1}{n}\right)^{5/2}, \left(\frac{1}{n}\right)^{5/3}, \left(\frac{1}{n}\right)^{7/2}, \left(\frac{1}{n}\right)^{7/3}, \left(\frac{1}{n}\right)^{7/6}, \left(\frac{1}{n}\right)^{8/3},$$

$\left(\frac{1}{n}\right)^{9/2}, \left(\frac{1}{n}\right)^{10/3}, \left(\frac{1}{n}\right)^{11/3}, \left(\frac{1}{n}\right)^{11/6}, \left(\frac{1}{n}\right)^{13/3}, \left(\frac{1}{n}\right)^{13/6},$
$\left(\frac{1}{n}\right)^{14/3}, \left(\frac{1}{n}\right)^{17/6}, \left(\frac{1}{n}\right)^{19/6}, \left(\frac{1}{n}\right)^{23/6}, \left(\frac{1}{n}\right)^{25/6}, \left(\frac{1}{n}\right)^{29/6},$
$\mathrm{O}\left(\frac{1}{n^5}\right)\}$

The mynewt function computes the Newton polygon of posFf

```
> newt1 := mynewt(posFf,m,n):
```

First Newton polygon, where no unknowns have been fixed

```
> P1 := pointplot(newt1,myoptionsLo,color=blue):
  display(P1,myview);
```



Here, we want to kill the element (0,0)

```
> getel(posFf,0,0);
```

$$-\kappa a + 2\kappa \quad \textbf{(2.4.2.6)}$$

Set a=2

```
> posFfa := expand(simplify(subs(a=2,posFf))) assuming
  n::posint,m::posint:
> newta := mynewt(posFfa,m,n):
```

All blue points have been eliminated, and only the red ones remain

```
> P1a := pointplot(newta,myoptionsLo,color=red):
  display(P1,P1a,myview);
```



b=0 is forced due to the term n^{-1/3}

```
> getel(posFfa,0,-1/3);
```

$$-\frac{\kappa b}{n^{1/3}} \tag{2.4.2.7}$$

set a=2, b=0

```
> posFfab := expand(simplify(subs(b=0,posFfa))) assuming
  n::posint,m::posint:
> newtab := mynewt(posFfab,m,n):
```

Now only the green points remain.

```
> P1ab := pointplot(newtab,myoptionsLo,color=green):
  display(P1,P1a,P1ab,myview);
```



at this point we find our choice for c, which we heuristically computed already before in Section 3

```
> csubs;
  factor(getel(posFfab,0,-2/3));#subs(csubs,%);
  factor(getel(posFfab,1,-5/3));#subs(csubs,%);
  factor(getel(posFfab,2,-5/3));#subs(csubs,%);
```

$$c = a1\, 2^{2/3}$$

$$\frac{\kappa \left(a1\, 2^{2/3} - c\right)}{n^{2/3}}$$

$$\frac{\kappa\, m \left(a1\, 2^{2/3} q_1 + 4\, a1\, 2^{2/3} q_2 + 3\, a1\, 2^{2/3} - c\, q_1 - c\right)}{n^{5/3}}$$

$$\frac{\kappa\, m^2 q_2 \left(a1\, 2^{2/3} - c\right)}{n^{5/3}} \quad \textbf{(2.4.2.8)}$$

set a=2, b=0, c=a1*2^(2/3)

```
> posFfabc := expand(simplify(subs(csubs,posFfab)))
  assuming n::posint,m::posint:
> newtabc := mynewt(posFfabc,m,n):
> P1abc := pointplot(newtabc,myoptionsLo,color=yellow):
  display(P1,P1a,P1ab,P1abc,myview);
```



Here we get q[2] and pterm

```
> factor(getel(posFfabc,0,-1));
  factor(getel(posFfabc,1,-4/3));
  solve({%%,%},{q[2],pterm});
```

$$-\frac{\kappa\,(pterm-2\,q_2-4)}{n}$$

$$\frac{4\,\lambda\,2^{1/3}\,m\,(2+3\,q_2)}{3\,n^{4/3}}$$

$$\left\{pterm=\frac{8}{3},\,q_2=-\frac{2}{3}\right\} \quad \textbf{(2.4.2.9)}$$

set a=2, b=0,c,pterm=8/3,q[2]=-2/3

```
> posFfabcd := expand(simplify(subs(pterm=8/3,q[2]=-2/3,
  posFfabc)) assuming n::posint,m::posint:
> newtabcd := mynewt(posFfabcd,m,n):
```

only the brown points remain

Now all points are strictly below n^{-1}

```
> P1abcd := pointplot(newtabcd,myoptionsLo,color=brown):
  display(P1,P1a,P1ab,P1abc,P1abcd,myview);
```



Here are the dominating corners and we see that we have to choose d=-1 to have a positive term;
note that we will see that the second term should be negative, as lambda=Ai' is negative for large m

```
> getel(posFfabcd,0,-7/6);
  getel(posFfabcd,3,-14/6);
  (14/6-7/6)/3;   #slope
```

$$-\frac{\kappa d}{n^{7/6}}$$

$$-\frac{16\, 2^{1/3} \lambda m^3}{9\, n^{7/3}}$$

$$\frac{7}{18} \quad \mathbf{(2.4.2.10)}$$

set a=2, b=0,c,pterm=8/3,q[2]=-2/3 and d=-1

```
> posFfabcde := expand(simplify(subs(d=-1,posFfabcd)))
  assuming n::posint,m::posint:
> newtabcde := mynewt(posFfabcde,m,n):
```

This is the final result, where only the solid diamonds are non-zero

```
> P1abcde := pointplot(newtabcde,myoptionsLo,symbol=
  soliddiamond,color=gray):
  display(P1,P1a,P1ab,P1abc,P1abcd,P1abcde,myview);
```



Plot the boundary and the slopes of the Newton polygon;
Note that we have already proved that there are now points above the blue dotted line

```
> LegendSize := size=[600,450]:
  P1dom1 := plot(-7/6-(7/18)*m,m=0..3,color=black,legend=
  [typeset("Slope: ", -7/18)],legendstyle=[location=right]
  ):
  P1dom2 := plot(-1/3-(2/3)*m,m=3..4,color=red,legend=
  [typeset("Slope: ", -2/3)],legendstyle=[location=right])
  :
  P1dom3a := plot(1-m,m=0..5,color=blue,linestyle=dot,
  legend=[typeset("Slope: ", -1)],legendstyle=[location=
  right]):
  P1all := display(P1,P1a,P1ab,P1abc,P1abcd,P1abcde,
  P1dom1,P1dom2,P1dom3a,myview,LegendSize);

  for i from 1 to 8 do
     P1dom3[i] := plot([[1+i/2,0],[1+i/2,-i/2]],color=
  blue,linestyle=dot):
  end:
  display(P1all,seq(P1dom3[i],i=1..8));
```



0
1
2
3
4
5
i
-1
-2
-3
-4
j
Slope: $-\frac{7}{18}$
Slope: $-\frac{2}{3}$
Slope: -1



This is the choice for SF

```
> SF(n);
  subs(a=2,b=0,csubs,pterm=8/3,d=-1,%);
```

$$a + \frac{b}{n^{1/3}} + \frac{c}{n^{2/3}} + \frac{pterm}{n} + \frac{d}{n^{7/6}}$$

$$2 + \frac{a1\ 2^{2/3}}{n^{2/3}} + \frac{8}{3\ n} - \frac{1}{n^{7/6}} \quad \textbf{(2.4.2.11)}$$

recall

```
> kaplam;
```

$$\mathrm{AiryAi}\left(\mathrm{AiryAiZeros}(1) + \frac{2^{1/3}\ m}{n^{1/3}}\right) = \kappa, \mathrm{AiryAi}\left(1, \mathrm{AiryAiZeros}(1) + \frac{2^{1/3}\ m}{n^{1/3}}\right) = \lambda \quad \textbf{(2.4.2.12)}$$

Look at the corners

Here we see the necessary choice: p[4]>2/9 (note that for m>n^(1/3) lambda is negative)

```
> getel(posFfabcde,0,-7/6);
  simplify(getel(posFfabcde,3,-14/6));
  simplify(getel(posFfabcd,4,-3));
```

```
simplify(getel(posFfabcd,4,-4));
```

$$\frac{\kappa}{n^{7/6}}$$

$$-\frac{16\,2^{1/3}\,\lambda\,m^3}{9\,n^{7/3}}$$

$$-\frac{34\,\kappa\,m^4}{9\,n^3}$$

$$-\frac{m^4\left(2128\,a1\,\lambda-868\,\kappa\,q_1+18857\,\kappa\right)}{1260\,n^4} \quad \textbf{(2.4.2.13)}$$

Now split the black dots into the contributions from Ai and Ai'

```
> indets(posFfabcd);
```

$$\left\{a1, d, \kappa, \lambda, m, n, q_0, q_1, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/6}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/3}}, \frac{1}{n^{11/6}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{7/6}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{3/2}}\right\} \quad \textbf{(2.4.2.14)}$$

Sanity check that there are no other contributions

```
> subs(kappa=0,lambda=0,posFfabcd);
```

$$0 \quad \textbf{(2.4.2.15)}$$

Extract the coefficients of kappa=Ai and lambda=Ai'
and treat then seperately

```
> posFk := coeff(posFfabcd,kappa):indets(%);
  posFl := coeff(posFfabcd,lambda):indets(%);
```

$$\left\{a1, d, m, n, q_0, q_1, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/6}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/3}}, \frac{1}{n^{11/6}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{7/6}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}\right\}$$

$$\left\{a1, d, m, n, q_0, q_1, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/6}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/3}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{3/2}}\right\} \quad \textbf{(2.4.2.16)}$$

We color the non-zero nodes of the lastNewton polygon into
red squared .... coefficients of kappa=Ai
blue diamonds ...... coefficients of lambda=Ai'

```
> newt4a := mynewt(posFk,m,n):
  newt4b := mynewt(posFl,m,n):
> P4a := pointplot(newt4a,labels=["m deg", "n deg"],
  symbolsize=25, symbol=box,color=red):
  P4b := pointplot(newt4b,labels=["m deg", "n deg"],
```

```
symbolsize=25, symbol=diamond, color=blue):
P1dom3s := plot(1-m,m=4..5,color=black):
display(P4a,P4b,P1dom1,P1dom2,P1dom3a,myview,
myoptionsLo,LegendSize);
```



red extremes of Newton polygon

```
> mnmaxR := getMaxNewt(Mord,newt4a):
  seq([i,mnmaxR[i]],i=0..Mord);
```

$$\left[0, -\frac{7}{6}\right], \left[1, -\frac{5}{3}\right], [2, -2], \left[3, -\frac{8}{3}\right], [4, -3], [5, -4] \tag{2.4.2.17}$$

These are the specific values at these points;
we see that we still have some degree in freedom: d and p[4]

```
> for i from 0 to Mord do
     i,factor(getel(posFk,i,mnmaxR[i]));
  end;
```

$$0, -\frac{d}{n^{7/6}}$$

$$1, -\frac{2\, a1\, 2^{2/3}\, m}{3\, n^{5/3}}$$

$$2, -\frac{41\,m^2}{9\,n^2}$$

$$3, -\frac{4\,2^{2/3}\,a1\,m^3}{3\,n^{8/3}}$$

$$4, -\frac{34\,m^4}{9\,n^3}$$

$$5, -\frac{62\,m^5}{135\,n^4} \qquad \textbf{(2.4.2.18)}$$

These are the slopes of the convex hull where the corners are given by the second sequence;
hence, in order to be positive when the slope > -1, we need to choose d>0, e.g. d=1;
note that p[4] is not important here, as the slope first slope -5/12 is less then -7/18;
and in the later regimes it will be dominated by the blue points.

```
> ls,li := getslopes(mnmaxR,Mord);
```

$$ls, li := \left[-\frac{5}{12}, -\frac{1}{2}, -1\right], [0, 2, 4, 5] \qquad \textbf{(2.4.2.19)}$$

```
> colors := [green,black,brown,blue,olive,red]:
  styles := [spacedot,solid, dash, dot, dashdot, longdash,
  spacedash]:
```

Draw the conveux hull in red

```
> for i from 1 to nops(ls) do
     ls[i];li[i];
     tt[i] := plot((mnmaxR[li[i]]-ls[i]*li[i])+ls[i]*m,m=
  li[i]..li[i+1],color=colors[i mod nops(colors)+1],
  linestyle=styles[i mod nops(styles)+1],legend=[typeset
  ("Slope: ", ls[i])],legendstyle=[location=right]):
  end:
  Pconvred := seq(tt[i],i=1..nops(ls)):
  #display(%);
> display(Pconvred,P4a,myview,myoptionsLo,LegendSize);
```



We continue with the blue diamonds, i.e., the coefficients of Ai'

```
> mnmaxB := getMaxNewt(Mord,newt4b):
  seq([i,mnmaxB[i]],i=0..Mord);
```

$$\left[0, -\frac{4}{3}\right], [1, -2], \left[2, -\frac{7}{3}\right], \left[3, -\frac{7}{3}\right], \left[4, -\frac{10}{3}\right], \left[5, -\frac{13}{3}\right] \quad (2.4.2.20)$$

```
> for i from 0 to Mord do
     i,factor(getel(posFl,i,mnmaxB[i]));
  end;
```

$$0, \frac{2^{1/3}\left(2\, q_1 - 1\right)}{n^{4/3}}$$

$$1, -\frac{8\, a1\, m}{9\, n^2}$$

$$2, \frac{2^{1/3}\, m^2\left(24\, q_1 - 31\right)}{9\, n^{7/3}}$$

$$3, -\frac{16\, 2^{1/3}\, m^3}{9\, n^{7/3}}$$

$$4, -\frac{10\, 2^{1/3}\, m^4}{9\, n^{10/3}}$$

$$5, -\frac{178\, 2^{1/3}\, m^5}{135\, n^{13/3}} \tag{2.4.2.21}$$

Again we derive the slopes of the convex hull and its corners;
note that if we choose q[1]=1/2 we eliminate the first term and decrease the slope.
This will be useful in the proof of Lemma 5.3, in the same way as it was used in the proof of Lemma 4.4.

```
> ls,li := getslopes(mnmaxB,Mord);
```

$$ls, li := \left[-\frac{1}{3}, -1\right], [0, 3, 5] \tag{2.4.2.22}$$

```
> colors := [green,olive,blue,black,brown,blue,red]:
  styles := [spacedot,solid, dot, dash, dashdot, longdash,
  spacedash]:
```

Draw the conveux hull in blue which still includes the term of order Theta(n^(-4/3))

```
> for i from 1 to nops(ls) do
     ls[i];li[i];
     tt[i] := plot((mnmaxB[li[i]]-ls[i]*li[i])+ls[i]*m,m=
  li[i]..li[i+1],color=colors[i mod nops(colors)+1],
  linestyle=styles[i mod nops(styles)+1],legend=[typeset
  ("Slope: ", ls[i])],legendstyle=[location=right]):
  end:
  Pconvblue := seq(tt[i],i=1..nops(ls)):
  #display(%);
> display(Pconvblue,P4b,myview,myoptionsLo,LegendSize);
```



We kill this term by setting q[1] = 1/2 and recompute the Newton polygons.
In the next picture the left-top blue point disappeared.
(note that the coefficients of lower order terms of posFk change as well, which is why we recompute them; however the picture will not change as the convex hull is not influenced by this change.)

```
> posFk2 := coeff(subs(q[1]=1/2,posFfabcde),kappa):indets
  (%);
  newt4a2 := mynewt(posFk2,m,n):
  P4a2 := pointplot(newt4a2,labels=["m deg", "n deg"],
  symbolsize=25, symbol=box, color=red):
  posFl2 := coeff(subs(q[1]=1/2,posFfabcde),lambda):indets
  (%);
  newt4b2 := mynewt(posFl2,m,n):
  P4b2 := pointplot(newt4b2,labels=["m deg", "n deg"],
  symbolsize=25, symbol=diamond, color=blue):
  display(P4a2,P4b2,P1dom1,P1dom2,P1dom3a,myview,
  myoptionsLo,LegendSize);
```

$$\left\{a1, m, n, q_0, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/6}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/3}}, \right.$$

$$\frac{1}{n^{13/6}}, \frac{1}{n^{11/3}}, \frac{1}{n^{11/6}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{7/6}},$$

$$\frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}\Bigg\}$$

$$\Bigg\{a1, m, n, q_0, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/6}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/3}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{3/2}}\Bigg\}$$



Recompute blue with q[1]=1/2

```
> mnmaxB := getMaxNewt(Mord,newt4b2):
  seq([i,mnmaxB[i]],i=0..Mord);
```

$$\left[0, -\frac{3}{2}\right], [1, -2], \left[2, -\frac{7}{3}\right], \left[3, -\frac{7}{3}\right], \left[4, -\frac{10}{3}\right], \left[5, -\frac{13}{3}\right] \quad (2.4.2.23)$$

```
> for i from 0 to Mord do
     i,factor(getel(posF12,i,mnmaxB[i]));
  end;
```

$$0, \frac{2^{1/3}}{n^{3/2}}$$

$$1, -\frac{8\,a1\,m}{9\,n^2}$$

$$2, -\frac{19\,2^{1/3}\,m^2}{9\,n^{7/3}}$$

$$3, -\frac{16\,2^{1/3}\,m^3}{9\,n^{7/3}}$$

$$4, -\frac{10\,2^{1/3}\,m^4}{9\,n^{10/3}}$$

$$5, -\frac{178\,2^{1/3}\,m^5}{135\,n^{13/3}} \quad \textbf{(2.4.2.24)}$$

And we get new slopes, yet at the same m powers given in the second sequence;
here we see that the term m^3 is negative, which will dominate in the regime when Ai' is negative.

```
> ls,li := getslopes(mnmaxB,Mord);
```

$$ls, li := \left[-\frac{5}{18}, -1\right], [0, 3, 5] \quad \textbf{(2.4.2.25)}$$

Draw the new conveux hull in blue

```
> for i from 1 to nops(ls) do
     ls[i];li[i];
     tt[i] := plot((mnmaxB[li[i]]-ls[i]*li[i])+ls[i]*m,m=
  li[i]..li[i+1],color=colors[i mod nops(colors)+1],
  linestyle=styles[i mod nops(styles)+1],legend=[typeset
  ("Slope: ", ls[i])],legendstyle=[location=right]):
  end:
  Pconvblue := seq(tt[i],i=1..nops(ls)):
  #display(%);
```

Plot the difference to before:
Only the solid circle on the top-left disappeared.

```
> P4bdiffshort := pointplot([0,-4/3],labels=["m deg", "n
  deg"], symbolsize=25, symbol=solidcircle, color=blue):
  display(Pconvblue,P4bdiffshort,P4b2,myview,myoptionsLo,
  LegendSize);
```



Relaxed Trees - Upper bound (Lemma 4.4)

Xhat

This is Y hat

```
> Xansatz := (n,m) -> (1-(4*m^2-3*m)/(6*n)+1/3*m^4/n^2)*
  AiryAi(a1+2^(1/3)*(m+1)/n^(1/3));
```

$$Xansatz := (n, m) \mapsto \left(1 - \frac{4\,m^2 - 3\,m}{6\,n} + \frac{m^4}{3\,n^2}\right) \mathrm{AiryAi}\left(a1 + \frac{2^{1/3}\,(m+1)}{n^{1/3}}\right) \tag{2.5.1.1}$$

```
> Sansatz := n -> 2 + c*n^(-2/3) + pterm/n + 1/(n^(7/6));
```

$$Sansatz := n \mapsto 2 + \frac{c}{n^{2/3}} + \frac{pterm}{n} + \frac{1}{n^{7/6}} \tag{2.5.1.2}$$

lower bound
(only difference to upper bound is missing factor (n-m-4)/(n-m-2) multiplied with the last

two terms;
as it is a lower bound, and we still want to prove positivity, we multiply the full equation with -1)

```
> posansatz := XX(n,m)*SS(n) - ( (n-m+2)/(n+m)*XX(n-1,m-1)
  + XX(n-1,m+1) );
```

$$posansatz := XX(n, m)\, SS(n) - \frac{(n-m+2)\, XX(n-1, m-1)}{n+m} - XX(n-1, m+1) \quad \textbf{(2.5.1.3)}$$

```
> posXS := map(simplify,subs(XX=Xansatz,SS=Sansatz,
  posansatz)):
```

For a large n this function of m seems to be positive

```
> Digits:=20:
  e1 := subs(csubs,pterm=8/3,a1=A1,posXS):
  N := 100000;
  M := floor(N^(1/2));
  P1 := plot([seq([mm,(subs(n=N,m=mm,e1))],mm=0..M)])
  :display(P1);
```

$$N := 100000$$

$$M := 316$$



Prove it

We start with the ansatz of Yhat in Lemma 5.3.

Recall the general ansatz

```
> facAiryUp*Airy(a1+2^(1/3)*(m+1)/n^(1/3));;
```

```
SF(n);
```

$$\left(1+\frac{m^4 p_4+m^3 p_3+m^2 p_2+m p_1+p_0}{n^2}+\frac{m^2 q_2+m q_1+q_0}{n}\right) Airy\left(a1+\frac{2^{1/3}(m+1)}{n^{1/3}}\right)$$

$$a+\frac{b}{n^{1/3}}+\frac{c}{n^{2/3}}+\frac{pterm}{n}+\frac{d}{n^{7/6}} \quad (2.5.2.1)$$

Substitute ansatz into sequence we want to be positive for large n and all m

```
> posF := map(expand, subs(XX=XFU,SS=SF,posansatz)):indets
  (%);
```

$$\left\{a, a1, b, c, d, \kappa, \lambda, m, n, pterm, p_0, p_1, p_2, p_3, p_4, q_0, q_1, q_2, \frac{1}{n^{8/3}}, \frac{1}{n^{7/3}}, \frac{1}{n^{7/6}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{2/3}}, \frac{1}{n^{1/3}}, \frac{1}{(n-1)^{1/3}}\right\} \quad (2.5.2.2)$$

The error terms are (to check, look at posFc)

```
> simplify(O((2^(1/3)*(m+1)/n^(1/3)-2^(1/3)*m/n^(1/3))
  ^ordAiUp));
  simplify(O((2^(1/3)*(m)/(n-1)^(1/3)-2^(1/3)*m/n^(1/3))
  ^ordAiUp));
  simplify(O((2^(1/3)*(m+2)/(n-1)^(1/3)-2^(1/3)*m/n^(1/3))
  ^ordAiUp));
```

$$O\left(\frac{64\, 2^{1/3}}{n^{19/3}}\right)$$

$$O\left(-\frac{64\, 2^{1/3} m^{19}\left((n-1)^{1/3}-n^{1/3}\right)^{19}}{(n-1)^{19/3} n^{19/3}}\right)$$

$$O\left(\frac{64\left(-m(n-1)^{1/3}+(m+2) n^{1/3}\right)^{19} 2^{1/3}}{n^{19/3}(n-1)^{19/3}}\right) \quad (2.5.2.3)$$

remove error terms

```
> posFd := convert(posF,polynom):
```

Next we fit the display size of the plots in the n (Nord) and m (Mord) direction.
(Note that everything up to ordAi is computed, but possibly not shown)

```
> Nord := -ordAiUp/3;
  Mord := floor(ordAiUp/3)+1;
  myview := view=[0..Mord,Nord..0]:
```

$$Nord := -\frac{19}{3}$$

$$Mord := 7 \quad (2.5.2.4)$$

Expand again with respect to n,
these are then our unknowns

```
> posFe := series(posFd,n=infinity,ceil(-Nord)+1):indets
  (%);
  posFf := convert(%%,polynom):
```

$$\left\{a, a1, b, c, d, \kappa, \lambda, m, n, pterm, p_0, p_1, p_2, p_3, p_4, q_0, q_1, q_2, \left(\frac{1}{n}\right)^{1/3}, \left(\frac{1}{n}\right)^{2/3}, \left(\frac{1}{n}\right)^{3/2}, \left(\frac{1}{n}\right)^{4/3}, \left(\frac{1}{n}\right)^{5/2}, \left(\frac{1}{n}\right)^{5/3}, \left(\frac{1}{n}\right)^{7/2}, \left(\frac{1}{n}\right)^{7/3}, \left(\frac{1}{n}\right)^{7/6},\right. \quad (2.5.2.5)$$

$$\left(\frac{1}{n}\right)^{8/3}, \left(\frac{1}{n}\right)^{9/2}, \left(\frac{1}{n}\right)^{10/3}, \left(\frac{1}{n}\right)^{11/2}, \left(\frac{1}{n}\right)^{11/3}, \left(\frac{1}{n}\right)^{11/6},$$
$$\left(\frac{1}{n}\right)^{13/2}, \left(\frac{1}{n}\right)^{13/3}, \left(\frac{1}{n}\right)^{13/6}, \left(\frac{1}{n}\right)^{14/3}, \left(\frac{1}{n}\right)^{16/3}, \left(\frac{1}{n}\right)^{17/3},$$
$$\left(\frac{1}{n}\right)^{17/6}, \left(\frac{1}{n}\right)^{19/3}, \left(\frac{1}{n}\right)^{19/6}, \left(\frac{1}{n}\right)^{20/3}, \left(\frac{1}{n}\right)^{23/6}, \left(\frac{1}{n}\right)^{25/6},$$
$$\left(\frac{1}{n}\right)^{29/6}, \left(\frac{1}{n}\right)^{31/6}, \left(\frac{1}{n}\right)^{35/6}, \left(\frac{1}{n}\right)^{37/6}, \left(\frac{1}{n}\right)^{41/6}, \mathrm{O}\left(\frac{1}{n^7}\right)\Big\}$$

The mynewt function computes the Newton polygon of posFf

```
> newt1 := mynewt(posFf,m,n):
```

First Newton polygon, where no unknowns have been fixed

```
> P1 := pointplot(newt1,myoptionsUp,color=blue):
  display(P1,myview);
```



Here, we want to kill the element (0,0)

```
> getel(posFf,0,0);
```

$$\kappa a - 2\kappa \tag{2.5.2.6}$$

Set a=2

```
> posFfa := expand(simplify(subs(a=2,posFf))) assuming
  n::posint,m::posint:
```

```
> newta := mynewt(posFfa,m,n):
```

All blue points have been eliminated, and only the red ones remain

```
> P1a := pointplot(newta,myoptionsUp,color=red):
  display(P1,P1a,myview);
```



b=0 is forced due to the term n^{-1/3)

```
> getel(posFfa,0,-1/3);
```

$$\frac{\kappa b}{n^{1/3}} \quad (2.5.2.7)$$

set a=2, b=0

```
> posFfab := expand(simplify(subs(b=0,posFfa))) assuming
  n::posint,m::posint:
> newtab := mynewt(posFfab,m,n):
```

Now only the green points remain.

```
> P1ab := pointplot(newtab,myoptionsUp,color=green):
  display(P1,P1a,P1ab,myview);
```



at this point we find our choice for c, which we heuristically computed already before in Section 3

```
> csubs;
  factor(getel(posFfab,0,-2/3));#subs(csubs,%);
  factor(getel(posFfab,1,-5/3));#subs(csubs,%);
  factor(getel(posFfab,2,-5/3));#subs(csubs,%);
```

$$c = a1\, 2^{2/3}$$

$$-\frac{\kappa\left(a1\, 2^{2/3} - c\right)}{n^{2/3}}$$

$$-\frac{\kappa\, m\left(a1\, 2^{2/3} q_1 + 4\, a1\, 2^{2/3} q_2 + 3\, a1\, 2^{2/3} - c\, q_1 - c\right)}{n^{5/3}}$$

$$-\frac{\kappa\, m^2 q_2\left(a1\, 2^{2/3} - c\right)}{n^{5/3}} \qquad \textbf{(2.5.2.8)}$$

set a=2, b=0, c=a1*2^(2/3)

```
> posFfabc := expand(simplify(subs(csubs,posFfab)))
  assuming n::posint,m::posint:
> newtabc := mynewt(posFfabc,m,n):
> P1abc := pointplot(newtabc,myoptionsUp,color=yellow):
  display(P1,P1a,P1ab,P1abc,myview);
```



Here we get q[2] and pterm

```
> factor(getel(posFfabc,0,-1));
  factor(getel(posFfabc,1,-4/3));
  solve({%%,%},{q[2],pterm});
```

$$\frac{\kappa\left(pterm-2\,q_2-4\right)}{n}$$

$$-\frac{4\,\lambda\,2^{1/3}\,m\left(2+3\,q_2\right)}{3\,n^{4/3}}$$

$$\left\{pterm=\frac{8}{3},\,q_2=-\frac{2}{3}\right\} \tag{2.5.2.9}$$

set a=2, b=0,c,pterm=8/3,q[2]=-2/3

```
> posFfabcd := expand(simplify(subs(pterm=8/3,q[2]=-2/3,
  posFfabc)) assuming n::posint,m::posint:
> newtabcd := mynewt(posFfabcd,m,n):
```

only the brown points remain
Now all points are strictly below n^{-1}

```
> P1abcd := pointplot(newtabcd,myoptionsUp,color=brown):
  display(P1,P1a,P1ab,P1abc,P1abcd,myview);
```



Here are the dominating corners and we see that we have to choose d=1 to have a positive term;
note that we will see that the second term should be negative, as lambda=Ai' is negative for large m,
hence here we will choose a p[4]>2/9 (see below in the decomposition into lambda and kappa contributions)

```
> getel(posFfabcd,0,-7/6);
  factor(getel(posFfabcd,3,-14/6));
  (14/6-7/6)/3;   #slope
```

$$\frac{\kappa d}{n^{7/6}}$$

$$-\frac{8\,2^{1/3}\lambda m^3\left(-2+9\,p_4\right)}{9\,n^{7/3}}$$

$$\frac{7}{18} \tag{2.5.2.10}$$

and continuing

```
> getel(posFfabcd,5,-20/6);
  (20/6-14/6)/2;    #slope
```

$$-\frac{8\,2^{1/3}\lambda m^5 p_4}{3\,n^{10/3}}$$

$$\frac{1}{2} \tag{2.5.2.11}$$

set a=2, b=0,c,pterm=8/3,q[2]=-2/3 and d=1

```
> posFfabcde := expand(simplify(subs(d=1,posFfabcd)))
  assuming n::posint,m::posint:
> newtabcde := mynewt(posFfabcde,m,n):
```

This is the final result, where only the solid diamonds are non-zero

```
> P1abcde := pointplot(newtabcde,myoptionsUp,symbol=
  soliddiamond,color=gray):
  display(P1,P1a,P1ab,P1abc,P1abcd,P1abcde,myview);
```



Plot the boundary and the slopes of the Newton polygon;
Note that we have already proved that there are now points above the blue dotted line

```
> LegendSize := size=[600,450]:
  P1dom1 := plot(-7/6-(7/18)*m,m=0..3,color=black,legend=
  [typeset("Slope: ", -7/18)],legendstyle=[location=right]
  ):
  P1dom2 := plot(-5/6-(1/2)*m,m=3..5,color=red,legend=
  [typeset("Slope: ", -1/2)],legendstyle=[location=right])
  :
  P1dom2b := plot(0-(2/3)*m,m=5..6,color=green,legend=
  [typeset("Slope: ", -2/3)],legendstyle=[location=right])
  :
  P1dom3a := plot(2-m,m=0..7,color=blue,linestyle=dot,
  legend=[typeset("Slope: ", -1)],legendstyle=[location=
  right]):
  P1all := display(P1,P1a,P1ab,P1abc,P1abcd,P1abcde,
  P1dom1,P1dom2,P1dom2b,P1dom3a,myview,LegendSize):
```

```
for i from 1 to 10 do
   P1dom3[i] := plot([[2+i/2,0],[2+i/2,-i/2]],color=
blue,linestyle=dot):
end:
display(P1all,seq(P1dom3[i],i=1..10));
```



This is the choice for SF

```
> SF(n);
  subs(a=2,b=0,csubs,pterm=8/3,d=1,%);
```

$$a+\frac{b}{n^{1/3}}+\frac{c}{n^{2/3}}+\frac{pterm}{n}+\frac{d}{n^{7/6}}$$

$$2+\frac{a1\,2^{2/3}}{n^{2/3}}+\frac{8}{3n}+\frac{1}{n^{7/6}} \quad \textbf{(2.5.2.12)}$$

recall

```
> kaplam;
```

$$\mathrm{AiryAi}\left(\mathrm{AiryAiZeros}(1)+\frac{2^{1/3}m}{n^{1/3}}\right)=\kappa,\ \mathrm{AiryAi}\left(1,\mathrm{AiryAiZeros}(1)+\frac{2^{1/3}m}{n^{1/3}}\right)=\lambda \quad \textbf{(2.5.2.13)}$$

Look at the corners
Here we see the necessary choice: p[4]>2/9 (note that for m>n^(1/3) lambda is negative)

```
> getel(posFfabcde,0,-7/6);
  simplify(getel(posFfabcde,3,-14/6));
  getel(posFfabcde,5,-20/6);
  getel(posFfabcde,6,-4);
  getel(posFfabcde,7,-5);
```

$$\frac{\kappa}{n^{7/6}}$$

$$-\frac{8\,2^{1/3}\,\lambda\,m^3\,(-2+9\,p_4)}{9\,n^{7/3}}$$

$$-\frac{8\,2^{1/3}\,\lambda\,m^5\,p_4}{3\,n^{10/3}}$$

$$-\frac{17\,\kappa\,m^6\,p_4}{3\,n^4}$$

$$-\frac{31\,\kappa\,m^7\,p_4}{45\,n^5} \quad \textbf{(2.5.2.14)}$$

Now split the black dots into the contributions from Ai and Ai'

```
> indets(posFfabcd);
```

$$\left\{a1, d, \kappa, \lambda, m, n, p_0, p_1, p_2, p_3, p_4, q_0, q_1, \frac{1}{n^{41/6}}, \frac{1}{n^{37/6}}, \frac{1}{n^{35/6}}, \frac{1}{n^{31/6}}, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{20/3}}, \frac{1}{n^{19/3}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/3}}, \frac{1}{n^{17/6}}, \frac{1}{n^{16/3}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/2}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/2}}, \frac{1}{n^{11/3}}, \frac{1}{n^{11/6}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{7/6}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{3/2}}\right\} \quad \textbf{(2.5.2.15)}$$

Sanity check that there are no other contributions

```
> subs(kappa=0,lambda=0,posFfabcd);
```

$$0 \quad \textbf{(2.5.2.16)}$$

Extract the coefficients of kappa=Ai and lambda=Ai'
and treat then seperately

```
> posFk := coeff(posFfabcd,kappa):indets(%);
  posFl := coeff(posFfabcd,lambda):indets(%);
```

$$\left\{a1, d, m, n, p_0, p_1, p_2, p_3, p_4, q_0, q_1, \frac{1}{n^{41/6}}, \frac{1}{n^{37/6}}, \frac{1}{n^{35/6}}, \frac{1}{n^{31/6}}, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{20/3}}, \frac{1}{n^{19/3}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/3}}, \frac{1}{n^{17/6}}, \frac{1}{n^{16/3}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/2}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/2}}, \frac{1}{n^{11/3}}, \frac{1}{n^{11/6}}, \frac{1}{n^{10/3}},\right.$$

$$\frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{7/6}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}\Big\}$$

$$\Big\{a1, d, m, n, p_0, p_1, p_2, p_3, p_4, q_0, q_1, \frac{1}{n^{41/6}}, \frac{1}{n^{37/6}}, \frac{1}{n^{35/6}}, \frac{1}{n^{31/6}}, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{20/3}}, \frac{1}{n^{19/3}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/3}}, \frac{1}{n^{17/6}}, \frac{1}{n^{16/3}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/2}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/2}}, \frac{1}{n^{11/3}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{3/2}}\Big\} \quad \textbf{(2.5.2.17)}$$

We color the non-zero nodes of the lastNewton polygon into
red squared .... coefficients of kappa=Ai
blue diamonds ...... coefficients of lambda=Ai'

```
> newt4a := mynewt(posFk,m,n):
  newt4b := mynewt(posFl,m,n):
> P4a := pointplot(newt4a,labels=["m deg", "n deg"],
  symbolsize=15, symbol=box,color=red):
  P4b := pointplot(newt4b,labels=["m deg", "n deg"],
  symbolsize=15, symbol=diamond, color=blue):
  P1dom3s := plot(1-m,m=4..5,color=black):
  display(P4a,P4b,P1dom1,P1dom2,P1dom2b,P1dom3a,myview,
  myoptionsUp,LegendSize);
```



red extremes of Newton polygon

```
> mnmaxR := getMaxNewt(Mord,newt4a):
  seq([i,mnmaxR[i]],i=0..Mord);
```

$$\left[0, -\frac{7}{6}\right], \left[1, -\frac{5}{3}\right], [2, -2], \left[3, -\frac{8}{3}\right], [4, -3], \left[5, -\frac{11}{3}\right], [6, -4], [7, -5] \quad (2.5.2.18)$$

These are the specific values at these points;
we see that we still have some degree in freedom: d and p[4]

```
> for i from 0 to Mord do
     i,factor(getel(posFk,i,mnmaxR[i]));
  end;
```

$$0, \frac{d}{n^{7/6}}$$

$$1, \frac{2\, a1\, 2^{2/3}\, m}{3\, n^{5/3}}$$

$$2, -\frac{m^2 (108\, p_4 - 41)}{9\, n^2}$$

$$3, -\frac{4\, 2^{2/3}\, a1\, m^3\, (6\, p_4 - 1)}{3\, n^{8/3}}$$

$$4, -\frac{2\, m^4\, (132\, p_4 - 17)}{9\, n^3}$$

$$5, -\frac{2\, 2^{2/3}\, a1\, m^5\, p_4}{n^{11/3}}$$

$$6, -\frac{17\, m^6\, p_4}{3\, n^4}$$

$$7, -\frac{31\, m^7\, p_4}{45\, n^5} \quad \textbf{(2.5.2.19)}$$

These are the slopes of the convex hull where the corners are given by the second sequence;
hence, in order to be positive when the slope > -1, we need to choose d>0, e.g. d=1;
note that p[4] is not important here, as the slope first slope -5/12 is less then -7/18;
and in the later regimes it will be dominated by the blue points.

```
> ls,li := getslopes(mnmaxR,Mord);
```

$$ls, li := \left[-\frac{5}{12}, -\frac{1}{2}, -1\right], [0, 2, 6, 7] \quad \textbf{(2.5.2.20)}$$

```
> colors := [green,black,brown,blue,olive,red]:
  styles := [spacedot,solid, dash, dot, dashdot, longdash,
  spacedash]:
```

Draw the conveux hull in red

```
> for i from 1 to nops(ls) do
     ls[i];li[i];
     tt[i] := plot((mnmaxR[li[i]]-ls[i]*li[i])+ls[i]*m,m=
  li[i]..li[i+1],color=colors[i mod nops(colors)+1],
  linestyle=styles[i mod nops(styles)+1],legend=[typeset
  ("Slope: ", ls[i])],legendstyle=[location=right]):
  end:
  Pconvred := seq(tt[i],i=1..nops(ls)):
  #display(%);
> display(Pconvred,P4a,myview,myoptionsUp,LegendSize);
```



We continue with the blue diamonds, i.e., the coefficients of Ai'

```
> mnmaxB := getMaxNewt(Mord,newt4b):
  seq([i,mnmaxB[i]],i=0..Mord);
```

$$\left[0, -\frac{4}{3}\right], [1, -2], \left[2, -\frac{7}{3}\right], \left[3, -\frac{7}{3}\right], \left[4, -\frac{10}{3}\right], \left[5, -\frac{10}{3}\right], \left[6, -\frac{13}{3}\right], \left[7, -\frac{16}{3}\right] \qquad (2.5.2.21)$$

```
> for i from 0 to Mord do
     i,factor(getel(posF1,i,mnmaxB[i]));
  end;
```

$$0, -\frac{2^{1/3}\left(2 q_1 - 1\right)}{n^{4/3}}$$

$$1, \frac{8\, a1\, m}{9\, n^2}$$

$$2, -\frac{2^{1/3} m^2 \left(54 p_3 + 108 p_4 + 24 q_1 - 31\right)}{9\, n^{7/3}}$$

$$3,\ -\frac{8\,2^{1/3}\,m^3\,(-2+9\,p_4)}{9\,n^{7/3}}$$

$$4,\ -\frac{2^{1/3}\,m^4\,(24\,p_3+135\,p_4-10)}{9\,n^{10/3}}$$

$$5,\ -\frac{8\,2^{1/3}\,m^5\,p_4}{3\,n^{10/3}}$$

$$6,\ -\frac{5\,2^{1/3}\,m^6\,p_4}{3\,n^{13/3}}$$

$$7,\ -\frac{89\,2^{1/3}\,m^7\,p_4}{45\,n^{16/3}} \qquad \textbf{(2.5.2.22)}$$

Again we derive the slopes of the convex hull and its corners;
note that if we choose q[1]=1/2 we eliminate the first term and decrease the slope.
This will be useful in the proof of Lemma 5.3, in the same way as it was used in the proof of Lemma 4.4.

```
> ls,li := getslopes(mnmaxB,Mord);
```

$$ls, li := \left[-\frac{1}{3}, -\frac{1}{2}, -1\right], [0, 3, 5, 7] \qquad \textbf{(2.5.2.23)}$$

```
> colors := [brown,olive,purple,blue,olive,red,black]:
  styles := [spacedot,solid, dash, dot, dashdot, longdash,
  spacedash]:
```

Draw the conveux hull in blue which still includes the term of order Theta(n^(-4/3))

```
> for i from 1 to nops(ls) do
     ls[i];li[i];
     tt[i] := plot((mnmaxB[li[i]]-ls[i]*li[i])+ls[i]*m,m=
  li[i]..li[i+1],color=colors[i mod nops(colors)+1],
  linestyle=styles[i mod nops(styles)+1],legend=[typeset
  ("Slope: ", ls[i])],legendstyle=[location=right]):
  end:
  Pconvblue := seq(tt[i],i=1..nops(ls)):
  #display(%);
> display(Pconvblue,P4b,myview,myoptionsUp,LegendSize);
```



We kill this term by setting q[1] = 1/2 and recompute the Newton polygons.
In the next picture the left-top blue point disappeared.
(note that the coefficients of lower order terms of posFk change as well, which is why we recompute them; however the picture will not change as the convex hull is not influenced by this change.)

```
> posFk2 := coeff(subs(q[1]=1/2,p[3]=0,p[2]=0,p[1]=0,p[0]=
  0,q[0]=0,posFfabcde),kappa):indets(%):
  newt4a2 := mynewt(posFk2,m,n):
  P4a2 := pointplot(newt4a2,labels=["m deg", "n deg"],
  symbolsize=15, symbol=box, color=red):
  posFl2 := coeff(subs(q[1]=1/2,p[3]=0,p[2]=0,p[1]=0,p[0]=
  0,q[0]=0,posFfabcde),lambda):indets(%):
  newt4b2 := mynewt(posFl2,m,n):
  P4b2 := pointplot(newt4b2,labels=["m deg", "n deg"],
  symbolsize=15, symbol=diamond, color=blue):
  display(P4a2,P4b2,P1dom1,P1dom2,P1dom2b,P1dom3a,myview,
  myoptionsUp,LegendSize);
```



Recompute blue with q[1]=1/2

```
> mnmaxB := getMaxNewt(Mord,newt4b2):
  seq([i,mnmaxB[i]],i=0..Mord);
```

$$\left[0, -\frac{3}{2}\right], [1, -2], \left[2, -\frac{7}{3}\right], \left[3, -\frac{7}{3}\right], \left[4, -\frac{10}{3}\right], \left[5, -\frac{10}{3}\right], \left[6, -\frac{13}{3}\right], \left[7, -\frac{16}{3}\right] \quad \textbf{(2.5.2.24)}$$

```
> for i from 0 to Mord do
     i,factor(getel(posFl2,i,mnmaxB[i]));
  end;
```

$$0, \frac{2^{1/3}}{n^{3/2}}$$

$$1, \frac{8\, a1\, m}{9\, n^2}$$

$$2, -\frac{2^{1/3}\, m^2\, (108\, p_4 - 19)}{9\, n^{7/3}}$$

$$3, -\frac{8\,2^{1/3}\,m^3\,(-2+9\,p_4)}{9\,n^{7/3}}$$
$$4, -\frac{5\,2^{1/3}\,m^4\,(-2+27\,p_4)}{9\,n^{10/3}}$$
$$5, -\frac{8\,2^{1/3}\,m^5\,p_4}{3\,n^{10/3}}$$
$$6, -\frac{5\,2^{1/3}\,m^6\,p_4}{3\,n^{13/3}}$$
$$7, -\frac{89\,2^{1/3}\,m^7\,p_4}{45\,n^{16/3}} \qquad \textbf{(2.5.2.25)}$$

And we get new slopes, yet at the same m powers given in the second sequence;
here we need that the term m^3 is negative as it will dominate in the regime when Ai' is negative,
therefore we have to choose p[4]>2/9 (note that this also makes m^5 and m^7 multiplied by Ai' positive)

```
> ls,li := getslopes(mnmaxB,Mord);
```

$$ls, li := \left[-\frac{5}{18}, -\frac{1}{2}, -1\right], [0, 3, 5, 7] \qquad \textbf{(2.5.2.26)}$$

Draw the new conveux hull in blue

```
> for i from 1 to nops(ls) do
     ls[i];li[i];
     tt[i] := plot((mnmaxB[li[i]]-ls[i]*li[i])+ls[i]*m,m=
  li[i]..li[i+1],color=colors[i mod nops(colors)+1],
  linestyle=styles[i mod nops(styles)+1],legend=[typeset
  ("Slope: ", ls[i])],legendstyle=[location=right]):
  end:
  Pconvblue := seq(tt[i],i=1..nops(ls)):
  #display(%);
```

Plot the difference to before:
Only the solid circle on the top-left disappeared.

```
> P4bdiffshort := pointplot([0,-4/3],labels=["m deg", "n
  deg"], symbolsize=15, symbol=solidcircle, color=blue):
  display(Pconvblue,P4bdiffshort,P4b2,myview,myoptionsUp,
  LegendSize);
```



Compacted Trees - Lower bound (Lemma 5.2)

Ytilde

This is Y tilde

```
> Xansatz := (n,m) -> (1-2*(m)^2/(3*n)+m/(4*n))* AiryAi
  (a1+2^(1/3)*(m+1)/n^(1/3));
```

$$Xansatz := (n, m) \mapsto \left(1 - \frac{2\,m^2}{3\,n} + \frac{m}{4\,n}\right) \mathrm{AiryAi}\left(a1 + \frac{2^{1/3}\,(m+1)}{n^{1/3}}\right) \quad \mathbf{(2.6.1.1)}$$

```
> Sansatz := n -> 2 + c*n^(-2/3) + pterm/n - 1/(n^(7/6));
```

$$Sansatz := n \mapsto 2 + \frac{c}{n^{2/3}} + \frac{pterm}{n} - \frac{1}{n^{7/6}} \quad \mathbf{(2.6.1.2)}$$

lower bound

```
> posansatz := -XX(n,m)*SS(n)*SS(n-1)*SS(n-2)
                   + (n-m+2)/(n+m)*XX(n-1,m-1)*SS(n-1)*SS
  (n-2)
                   + (n-m-2)/(n-m)*XX(n-1,m+1)*SS(n-1)*SS
```

```
(n-2)
                         + (n-m-4)/(n-m-2)*( 2/(n-m)*XX(n-2,m+2)*
SS(n-2) + 2/(n+m)*XX(n-3,m+1) )
                         ;
```

$$posansatz := -XX(n,m)\,SS(n)\,SS(n-1)\,SS(n-2) + \frac{(n-m+2)\,XX(n-1,m-1)\,SS(n-1)\,SS(n-2)}{n+m} + \frac{(n-m-2)\,XX(n-1,m+1)\,SS(n-1)\,SS(n-2)}{n-m} + \frac{1}{n-m-2}\left((n-m-4)\left(\frac{2\,XX(n-2,m+2)\,SS(n-2)}{n-m} + \frac{2\,XX(n-3,m+1)}{n+m}\right)\right) \quad \textbf{(2.6.1.3)}$$

```
> posXS := map(simplify,subs(XX=Xansatz,SS=Sansatz,
  posansatz)):
```

For a large n this function of m seems to be positive

```
> Digits:=20:
  e1 := subs(csubs,pterm=13/6,a1=A1,posXS):
  N := 100000;
  M := floor(N^(1/2));
  P1 := plot([seq([mm,(subs(n=N,m=mm,e1))],mm=0..M)])
  :display(P1);
```

$$N := 100000$$

$$M := 316$$



Prove it

We start with the ansatz of Ytilde in Lemma 5.2.

Recall the general ansatz

```
> facAiryLo*Airy(a1+2^(1/3)*(m+1)/n^(1/3));
  SF(n);
```

$$\left(1+\frac{m^2 q_2+m q_1+q_0}{n}\right) \mathit{Airy}\left(a1+\frac{2^{1/3}(m+1)}{n^{1/3}}\right)$$

$$a+\frac{b}{n^{1/3}}+\frac{c}{n^{2/3}}+\frac{pterm}{n}+\frac{d}{n^{7/6}} \quad \textbf{(2.6.2.1)}$$

Substitute ansatz into the sequence we want to be positive for large n and all m

```
> posF := map(expand, subs(XX=XFL,SS=SF,posansatz)):indets
  (%);
```

$$\left\{a, a1, b, c, d, \kappa, \lambda, m, n, pterm, q_0, q_1, q_2, \frac{1}{n^{7/6}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{2/3}}, \frac{1}{n^{1/3}}, \frac{1}{(n-3)^{1/3}}, \frac{1}{(n-2)^{7/6}}, \frac{1}{(n-2)^{2/3}}, \frac{1}{(n-2)^{1/3}}, \frac{1}{(n-1)^{7/6}}, \frac{1}{(n-1)^{2/3}}, \frac{1}{(n-1)^{1/3}}\right\} \quad \textbf{(2.6.2.2)}$$

The error terms are (to check, expand posF)

```
> simplify(O((2^(1/3)*(m+1)/n^(1/3)-2^(1/3)*m/n^(1/3))
  ^ordAiLo));
  simplify(O((2^(1/3)*(m)/(n-1)^(1/3)-2^(1/3)*m/n^(1/3))
  ^ordAiLo));
  simplify(O((2^(1/3)*(m+2)/(n-1)^(1/3)-2^(1/3)*m/n^(1/3))
  ^ordAiLo));
```

$$\mathrm{O}\left(\frac{16\, 2^{1/3}}{n^{13/3}}\right)$$

$$\mathrm{O}\left(-\frac{16\, 2^{1/3} m^{13} \left((n-1)^{1/3}-n^{1/3}\right)^{13}}{(n-1)^{13/3} n^{13/3}}\right)$$

$$\mathrm{O}\left(\frac{16\left(-m (n-1)^{1/3}+n^{1/3}(m+2)\right)^{13} 2^{1/3}}{n^{13/3}(n-1)^{13/3}}\right) \quad \textbf{(2.6.2.3)}$$

remove error terms

```
> posFd := convert(posF,polynom):
```

Next we fit the display size of the plots in the n (Nord) and m (Mord) direction.
(Note that everything up to ordAi is computed, but possibly not shown)

```
> Nord := -ordAiLo/3;
  Mord := floor(ordAiLo/3)+1;
  myview := view=[0..Mord,Nord..0]:
```

$$\mathit{Nord} := -\frac{13}{3}$$

$$\mathit{Mord} := 5 \quad \textbf{(2.6.2.4)}$$

Expand again with respect to n,
these are then our unknowns

```
> posFe := series(posFd,n=infinity,ceil(-Nord)+1):indets
  (%);
  posFf := convert(%%,polynom):
```

(2.6.2.5)

$$\left\{a, a1, b, c, d, \kappa, \lambda, m, n, pterm, q_0, q_1, q_2, \left(\frac{1}{n}\right)^{1/3}, \left(\frac{1}{n}\right)^{2/3}, \left(\frac{1}{n}\right)^{3/2}, \left(\frac{1}{n}\right)^{4/3}, \left(\frac{1}{n}\right)^{5/2}, \left(\frac{1}{n}\right)^{5/3}, \left(\frac{1}{n}\right)^{7/2}, \left(\frac{1}{n}\right)^{7/3}, \left(\frac{1}{n}\right)^{7/6}, \left(\frac{1}{n}\right)^{8/3}, \left(\frac{1}{n}\right)^{9/2}, \left(\frac{1}{n}\right)^{10/3}, \left(\frac{1}{n}\right)^{11/3}, \left(\frac{1}{n}\right)^{11/6}, \left(\frac{1}{n}\right)^{13/3}, \left(\frac{1}{n}\right)^{13/6}, \left(\frac{1}{n}\right)^{14/3}, \left(\frac{1}{n}\right)^{17/6}, \left(\frac{1}{n}\right)^{19/6}, \left(\frac{1}{n}\right)^{23/6}, \left(\frac{1}{n}\right)^{25/6}, \left(\frac{1}{n}\right)^{29/6}, \mathrm{O}\left(\frac{1}{n^5}\right)\right\} \quad \mathbf{(2.6.2.5)}$$

The mynewt function computes the Newton polygon of posFf

```
> newt1 := mynewt(posFf,m,n):
```

First Newton polygon, where no unknowns have been fixed

```
> P1 := pointplot(newt1,myoptionsLo,color=blue):
  display(P1,myview);
```



Here, we want to kill the element (0,0)

```
> getel(posFf,0,0);
```

$$-\kappa a^3 + 2 \kappa a^2 \quad \mathbf{(2.6.2.6)}$$

Set a=2

```
> posFfa := expand(simplify(subs(a=2,posFf))) assuming
```

```
n::posint,m::posint:
> newta := mynewt(posFfa,m,n):
```

All blue points have been eliminated, and only the red ones remain

```
> P1a := pointplot(newta,myoptionsLo,color=red):
  display(P1,P1a,myview);
```



b=0 is forced due to the term n^{-1/3)

```
> getel(posFfa,0,-1/3);
```

$$-\frac{4\,\kappa\,b}{n^{1/3}} \qquad \textbf{(2.6.2.7)}$$

set a=2, b=0

```
> posFfab := expand(simplify(subs(b=0,posFfa))) assuming
  n::posint,m::posint:
> newtab := mynewt(posFfab,m,n):
```

Now only the green points remain.

```
> P1ab := pointplot(newtab,myoptionsLo,color=green):
  display(P1,P1a,P1ab,myview);
```



at this point we find our choice for c, which we heuristically computed already before in Section 3

```
> csubs;
  factor(getel(posFfab,0,-2/3));#subs(csubs,%);
  factor(getel(posFfab,1,-5/3));#subs(csubs,%);
  factor(getel(posFfab,2,-5/3));#subs(csubs,%);
```

$$c = a1\, 2^{2/3}$$

$$\frac{4\,\kappa\left(a1\, 2^{2/3} - c\right)}{n^{2/3}}$$

$$\frac{4\,\kappa\, m\left(a1\, 2^{2/3}\, q_1 + 4\, a1\, 2^{2/3}\, q_2 + 3\, a1\, 2^{2/3} - c\, q_1 - c\right)}{n^{5/3}}$$

$$\frac{4\,\kappa\, m^2\, q_2\left(a1\, 2^{2/3} - c\right)}{n^{5/3}} \qquad \textbf{(2.6.2.8)}$$

set a=2, b=0, c=a1*2^(2/3)

```
> posFfabc := expand(simplify(subs(csubs,posFfab)))
  assuming n::posint,m::posint:
> newtabc := mynewt(posFfabc,m,n):
> P1abc := pointplot(newtabc,myoptionsLo,color=yellow):
  display(P1,P1a,P1ab,P1abc,myview);
```



Here we get q[2] and pterm

```
> factor(getel(posFfabc,0,-1));
  factor(getel(posFfabc,1,-4/3));
  solve({%%,%},{q[2],pterm});
```

$$-\frac{2\,\kappa\,(2\,pterm-4\,q_2-7)}{n}$$

$$\frac{16\,\lambda\,2^{1/3}\,m\,(2+3\,q_2)}{3\,n^{4/3}}$$

$$\left\{pterm=\frac{13}{6},\,q_2=-\frac{2}{3}\right\} \tag{2.6.2.9}$$

set a=2, b=0,c,pterm=13/6,q[2]=-2/3

```
> posFfabcd := expand(simplify(subs(pterm=13/6,q[2]=-2/3,
  posFfabc))) assuming n::posint,m::posint:
> newtabcd := mynewt(posFfabcd,m,n):
```

only the brown points remain
Now all points are strictly below n^{-1}

```
> P1abcd := pointplot(newtabcd,myoptionsLo,color=brown):
  display(P1,P1a,P1ab,P1abc,P1abcd,myview);
```



Here are the dominating corners and we see that we have to choose d=-1 to have a positive term;
note that we will see that the second term should be negative, as lambda=Ai' is negative for large m

```
> getel(posFfabcd,0,-7/6);
  getel(posFfabcd,3,-14/6);
  (14/6-7/6)/3;   #slope
```

$$-\frac{4\,\kappa\,d}{n^{7/6}}$$

$$-\frac{64\,2^{1/3}\,\lambda\,m^3}{9\,n^{7/3}}$$

$$\frac{7}{18} \quad \textbf{(2.6.2.10)}$$

and continuing

```
> getel(posFfabcd,5,-20/6);
  (20/6-14/6)/2;    #slope
```

$$0$$

$$\frac{1}{2} \quad \textbf{(2.6.2.11)}$$

set a=2, b=0,c,pterm=13/6,q[2]=-2/3 and d=-1

```
> posFfabcde := expand(simplify(subs(d=-1,posFfabcd)))
  assuming n::posint,m::posint:
> newtabcde := mynewt(posFfabcde,m,n):
```

This is the final result, where only the solid diamonds are non-zero

```
> P1abcde := pointplot(newtabcde,myoptionsLo,symbol=
  soliddiamond,color=gray):
  display(P1,P1a,P1ab,P1abc,P1abcd,P1abcde,myview);
```



Plot the boundary and the slopes of the Newton polygon;
Note that we have already proved that there are now points above the blue dotted line

```
> LegendSize := size=[600,450]:
  P1dom1 := plot(-7/6-(7/18)*m,m=0..3,color=black,legend=
  [typeset("Slope: ", -7/18)],legendstyle=[location=right]
  ):
  P1dom2 := plot(-1/3-(2/3)*m,m=3..4,color=red,legend=
  [typeset("Slope: ", -2/3)],legendstyle=[location=right])
  :
  P1dom3a := plot(1-m,m=0..5,color=blue,linestyle=dot,
  legend=[typeset("Slope: ", -1)],legendstyle=[location=
  right]):
  P1all := display(P1,P1a,P1ab,P1abc,P1abcd,P1abcde,
  P1dom1,P1dom2,P1dom3a,myview,LegendSize);

  for i from 1 to 8 do
     P1dom3[i] := plot([[1+i/2,0],[1+i/2,-i/2]],color=
  blue,linestyle=dot):
  end:
  display(P1all,seq(P1dom3[i],i=1..8));
```



0
1
2
3
4
5
i
-1
-2
-3
-4
j
Slope: $-\frac{7}{18}$
Slope: $-\frac{2}{3}$
Slope: -1



This is the choice for SF

```
> SF(n);
  subs(a=2,b=0,csubs,pterm=13/6,d=-1,%);
```

$$a+\frac{b}{n^{1/3}}+\frac{c}{n^{2/3}}+\frac{pterm}{n}+\frac{d}{n^{7/6}}$$

$$2+\frac{a1\,2^{2/3}}{n^{2/3}}+\frac{13}{6\,n}-\frac{1}{n^{7/6}} \qquad \textbf{(2.6.2.12)}$$

recall

```
> kaplam;
```

$$\mathrm{AiryAi}\left(\mathrm{AiryAiZeros}(1)+\frac{2^{1/3}\,m}{n^{1/3}}\right)=\kappa,\ \mathrm{AiryAi}\left(1,\ \mathrm{AiryAiZeros}(1)+\frac{2^{1/3}\,m}{n^{1/3}}\right)=\lambda \qquad \textbf{(2.6.2.13)}$$

Look at the corners

Here we see the necessary choice: p[4]>2/9 (note that for m>n^(1/3) lambda is negative)

```
> getel(posFfabcde,0,-7/6);
  simplify(getel(posFfabcde,3,-14/6));
  simplify(getel(posFfabcd,4,-3));
```

```
simplify(getel(posFfabcd,4,-4));
simplify(getel(posFfabcd,6,-5));
```

$$\frac{4\,\kappa}{n^{7/6}}$$

$$-\frac{64\,2^{1/3}\,\lambda\,m^3}{9\,n^{7/3}}$$

$$-\frac{136\,\kappa\,m^4}{9\,n^3}$$

$$\frac{2604\left(q_1-\frac{3751}{84}\right)m^4\,\kappa-14784\,a1\,\lambda\,m^4}{945\,n^4}$$

$$0 \qquad \textbf{(2.6.2.14)}$$

Now split the black dots into the contributions from Ai and Ai'

```
> indets(posFfabcd);
```

$$\left\{a1, d, \kappa, \lambda, m, n, q_0, q_1, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/6}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/3}}, \frac{1}{n^{11/6}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{7/6}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{3/2}}\right\} \qquad \textbf{(2.6.2.15)}$$

Sanity check that there are no other contributions

```
> subs(kappa=0,lambda=0,posFfabcd);
```

$$0 \qquad \textbf{(2.6.2.16)}$$

Extract the coefficients of kappa=Ai and lambda=Ai'
and treat then seperately

```
> posFk  := coeff(posFfabcd,kappa):indets(%);
  posFl  := coeff(posFfabcd,lambda):indets(%);
```

$$\left\{a1, d, m, n, q_0, q_1, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/6}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/3}}, \frac{1}{n^{11/6}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{7/6}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}\right\}$$

$$\left\{a1, d, m, n, q_0, q_1, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/6}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/3}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{3/2}}\right\} \qquad \textbf{(2.6.2.17)}$$

We color the non-zero nodes of the lastNewton polygon into
red squared .... coefficients of kappa=Ai
blue diamonds ...... coefficients of lambda=Ai'

```
> newt4a  := mynewt(posFk,m,n):
  newt4b  := mynewt(posFl,m,n):
```

```
> P4a := pointplot(newt4a,labels=["m deg", "n deg"],
  symbolsize=25, symbol=box,color=red):
  P4b := pointplot(newt4b,labels=["m deg", "n deg"],
  symbolsize=25, symbol=diamond, color=blue):
  P1dom3s := plot(1-m,m=4..5,color=black):
  display(P4a,P4b,P1dom1,P1dom2,P1dom3a,myview,
  myoptionsLo,LegendSize);
```



red extremes of Newton polygon

```
> mnmaxR := getMaxNewt(Mord,newt4a):
  seq([i,mnmaxR[i]],i=0..Mord);
```

$$\left[0, -\frac{7}{6}\right], \left[1, -\frac{5}{3}\right], [2, -2], \left[3, -\frac{8}{3}\right], [4, -3], [5, -4] \quad \textbf{(2.6.2.18)}$$

These are the specific values at these points;
we see that we still have some degree in freedom: d and p[4]

```
> for i from 0 to Mord do
     i,factor(getel(posFk,i,mnmaxR[i]));
  end;
```

$$0, -\frac{4\,d}{n^{7/6}}$$

$$1,\ -\frac{8\,a1\,2^{2/3}\,m}{3\,n^{5/3}}$$

$$2,\ -\frac{164\,m^2}{9\,n^2}$$

$$3,\ -\frac{16\,2^{2/3}\,a1\,m^3}{3\,n^{8/3}}$$

$$4,\ -\frac{136\,m^4}{9\,n^3}$$

$$5,\ -\frac{248\,m^5}{135\,n^4} \quad \textbf{(2.6.2.19)}$$

These are the slopes of the convex hull where the corners are given by the second sequence;
hence, in order to be positive when the slope > -1, we need to choose d>0, e.g. d=1;
note that p[4] is not important here, as the slope first slope -5/12 is less then -7/18;
and in the later regimes it will be dominated by the blue points.

```
> ls,li := getslopes(mnmaxR,Mord);
```

$$ls, li := \left[-\frac{5}{12}, -\frac{1}{2}, -1\right], [0, 2, 4, 5] \quad \textbf{(2.6.2.20)}$$

```
> colors := [green,black,brown,blue,olive,red]:
  styles := [spacedot,solid, dash, dot, dashdot, longdash,
  spacedash]:
```

Draw the conveux hull in red

```
> for i from 1 to nops(ls) do
     ls[i];li[i];
     tt[i] := plot((mnmaxR[li[i]]-ls[i]*li[i])+ls[i]*m,m=
  li[i]..li[i+1],color=colors[i mod nops(colors)+1],
  linestyle=styles[i mod nops(styles)+1],legend=[typeset
  ("Slope: ", ls[i])],legendstyle=[location=right]):
  end:
  Pconvred := seq(tt[i],i=1..nops(ls)):
  #display(%);
> display(Pconvred,P4a,myview,myoptionsLo,LegendSize);
```



We continue with the blue diamonds, i.e., the coefficients of Ai'

```
> mnmaxB := getMaxNewt(Mord,newt4b):
  seq([i,mnmaxB[i]],i=0..Mord);
```

$$\left[0, -\frac{4}{3}\right], [1, -2], \left[2, -\frac{7}{3}\right], \left[3, -\frac{7}{3}\right], \left[4, -\frac{10}{3}\right], \left[5, -\frac{13}{3}\right] \quad \textbf{(2.6.2.21)}$$

```
> for i from 0 to Mord do
     i,factor(getel(posFl,i,mnmaxB[i]));
  end;
```

$$0, \frac{2\, 2^{1/3} \left(4\, q_1 - 1\right)}{n^{4/3}}$$

$$1, -\frac{32\, a1\, m}{9\, n^2}$$

$$2, \frac{8\, 2^{1/3}\, m^2 \left(-17 + 12\, q_1\right)}{9\, n^{7/3}}$$

$$3, -\frac{64\, 2^{1/3}\, m^3}{9\, n^{7/3}}$$

$$4, -\frac{40\ 2^{1/3}\ m^4}{9\ n^{10/3}}$$

$$5, -\frac{712\ 2^{1/3}\ m^5}{135\ n^{13/3}} \qquad \textbf{(2.6.2.22)}$$

Again we derive the slopes of the convex hull and its corners;
note that if we choose q[1]=1/4 we eliminate the first term and decrease the slope.
This will be useful in the proof of Lemma 5.3, in the same way as it was used in the proof of Lemma 4.4.

```
> ls,li := getslopes(mnmaxB,Mord);
```

$$ls, li := \left[-\frac{1}{3}, -1\right], [0, 3, 5] \qquad \textbf{(2.6.2.23)}$$

```
> colors := [green,olive,blue,black,brown,blue,red]:
  styles := [spacedot,solid, dot, dash, dashdot, longdash,
  spacedash]:
```

Draw the conveux hull in blue which still includes the term of order Theta(n^(-4/3))

```
> for i from 1 to nops(ls) do
     ls[i];li[i];
     tt[i] := plot((mnmaxB[li[i]]-ls[i]*li[i])+ls[i]*m,m=
  li[i]..li[i+1],color=colors[i mod nops(colors)+1],
  linestyle=styles[i mod nops(styles)+1],legend=[typeset
  ("Slope: ", ls[i])],legendstyle=[location=right]):
  end:
  Pconvblue := seq(tt[i],i=1..nops(ls)):
  #display(%);
> display(Pconvblue,P4b,myview,myoptionsLo,LegendSize);
```



We kill this term by setting q[1] = 1/4 and recompute the Newton polygons.
In the next picture the left-top blue point disappeared.
(note that the coefficients of lower order terms of posFk change as well, which is why we recompute them; however the picture will not change as the convex hull is not influenced by this change.)

```
> posFk2 := coeff(subs(q[1]=1/4,posFfabcde),kappa):indets
  (%);
  newt4a2 := mynewt(posFk2,m,n):
  P4a2 := pointplot(newt4a2,labels=["m deg", "n deg"],
  symbolsize=25, symbol=box, color=red):
  posFl2 := coeff(subs(q[1]=1/4,posFfabcde),lambda):indets
  (%);
  newt4b2 := mynewt(posFl2,m,n):
  P4b2 := pointplot(newt4b2,labels=["m deg", "n deg"],
  symbolsize=25, symbol=diamond, color=blue):
  display(P4a2,P4b2,P1dom1,P1dom2,P1dom3a,myview,
  myoptionsLo,LegendSize);
```

$$\left\{a1, m, n, q_0, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/6}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/3}}, \right.$$

$$\frac{1}{n^{13/6}}, \frac{1}{n^{11/3}}, \frac{1}{n^{11/6}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{7/6}},$$

$$\frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}\}$$

$$\left\{a1, m, n, q_0, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/6}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/3}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{3/2}}\right\}$$



Recompute blue with q[1]=1/4

```
> mnmaxB := getMaxNewt(Mord,newt4b2):
  seq([i,mnmaxB[i]],i=0..Mord);
```

$$\left[0, -\frac{3}{2}\right], [1, -2], \left[2, -\frac{7}{3}\right], \left[3, -\frac{7}{3}\right], \left[4, -\frac{10}{3}\right], \left[5, -\frac{13}{3}\right] \quad \textbf{(2.6.2.24)}$$

```
> for i from 0 to Mord do
     i,factor(getel(posFl2,i,mnmaxB[i]));
  end;
```

$$0, \frac{4\, 2^{1/3}}{n^{3/2}}$$

$$1, -\frac{32\, a1\, m}{9\, n^2}$$

$$2, -\frac{112\, 2^{1/3}\, m^2}{9\, n^{7/3}}$$

$$3, -\frac{64\, 2^{1/3}\, m^3}{9\, n^{7/3}}$$

$$4, -\frac{40\, 2^{1/3}\, m^4}{9\, n^{10/3}}$$

$$5, -\frac{712\, 2^{1/3}\, m^5}{135\, n^{13/3}} \qquad \textbf{(2.6.2.25)}$$

And we get new slopes, yet at the same m powers given in the second sequence;
here we need that the term m^3 is negative as it will dominate in the regime when Ai' is negative,
therefore we have to choose p[4]>2/9 (note that this also makes m^5 and m^7 multiplied by Ai' positive)

```
> ls,li := getslopes(mnmaxB,Mord);
```

$$ls, li := \left[-\frac{5}{18}, -1\right], [0, 3, 5] \qquad \textbf{(2.6.2.26)}$$

Draw the new conveux hull in blue

```
> for i from 1 to nops(ls) do
     ls[i];li[i];
     tt[i] := plot((mnmaxB[li[i]]-ls[i]*li[i])+ls[i]*m,m=
  li[i]..li[i+1],color=colors[i mod nops(colors)+1],
  linestyle=styles[i mod nops(styles)+1],legend=[typeset
  ("Slope: ", ls[i])],legendstyle=[location=right]):
  end:
  Pconvblue := seq(tt[i],i=1..nops(ls)):
  #display(%);
```

Plot the difference to before:
Only the solid circle on the top-left disappeared.

```
> P4bdiffshort := pointplot([0,-4/3],labels=["m deg", "n
  deg"], symbolsize=25, symbol=solidcircle, color=blue):
  display(Pconvblue,P4bdiffshort,P4b2,myview,myoptionsLo,
  LegendSize);
```



Compacted Trees - Upper bound (Lemma 5.3)

Yhat

This is Y hat

```
> Xansatz := (n,m) -> (1-2*m^2/(3*n)+m/(4*n)+1/3*m^4/n^2)*
  AiryAi(a1+2^(1/3)*(m+1)/n^(1/3));
```

$$Xansatz := (n, m) \mapsto \left(1 - \frac{2\, m^2}{3\, n} + \frac{m}{4\, n} + \frac{m^4}{3\, n^2}\right) \mathrm{AiryAi}\left(a1 + \frac{2^{1/3}\, (m+1)}{n^{1/3}}\right) \quad \textbf{(2.7.1.1)}$$

```
> Sansatz := n -> 2 + c*n^(-2/3) + pterm/n + 1/(n^(7/6));
```

$$Sansatz := n \mapsto 2 + \frac{c}{n^{2/3}} + \frac{pterm}{n} + \frac{1}{n^{7/6}} \quad \textbf{(2.7.1.2)}$$

lower bound
(only difference to upper bound is missing factor (n-m-4)/(n-m-2) multiplied with the last

two terms;
as it is a lower bound, and we still want to prove positivity, we multiply the full equation with -1)

```
> posansatz := -(-XX(n,m)*SS(n)*SS(n-1)*SS(n-2)
                 + (n-m+2)/(n+m)*XX(n-1,m-1)*SS(n-1)*SS
  (n-2)
                 + (n-m-2)/(n-m)*XX(n-1,m+1)*SS(n-1)*SS
  (n-2)
                 + 2/(n-m)*XX(n-2,m+2)*SS(n-2)
                 + 2/(n+m)*XX(n-3,m+1)
                 + 4/(n+m)/(n+m-2)*XX(n-3,m-1)

                 );
```

$$posansatz := XX(n,m)\,SS(n)\,SS(n-1)\,SS(n-2) - \frac{(n-m+2)\,XX(n-1,m-1)\,SS(n-1)\,SS(n-2)}{n+m} - \frac{(n-m-2)\,XX(n-1,m+1)\,SS(n-1)\,SS(n-2)}{n-m} - \frac{2\,XX(n-2,m+2)\,SS(n-2)}{n-m} - \frac{2\,XX(n-3,m+1)}{n+m} - \frac{4\,XX(n-3,m-1)}{(n+m)\,(n+m-2)} \quad \textbf{(2.7.1.3)}$$

```
> posXS := map(simplify,subs(XX=Xansatz,SS=Sansatz,
  posansatz)):
```

For a large n this function of m seems to be positive

```
> Digits:=20:
  e1 := subs(csubs,pterm=13/6,a1=A1,posXS):
  N := 100000;
  M := floor(N^(1/2));
  P1 := plot([seq([mm,(subs(n=N,m=mm,e1))],mm=0..M)])
  :display(P1);
```

$$N := 100000$$
$$M := 316$$



Prove it

We start with the ansatz of Yhat in Lemma 5.3.

Recall the general ansatz

```
> facAiryUp*Airy(a1+2^(1/3)*(m+1)/n^(1/3));;
  SF(n);
```

$$\left(1+\frac{m^4 p_4+m^3 p_3+m^2 p_2+m p_1+p_0}{n^2}+\frac{m^2 q_2+m q_1+q_0}{n}\right) Airy\left(a1+\frac{2^{1/3}(m+1)}{n^{1/3}}\right)$$

$$a+\frac{b}{n^{1/3}}+\frac{c}{n^{2/3}}+\frac{pterm}{n}+\frac{d}{n^{7/6}} \tag{2.7.2.1}$$

Substitute ansatz into sequence we want to be positive for large n and all m

```
> posF := map(expand, subs(XX=XFU,SS=SF,posansatz)):indets
  (%);
```

$$\left\{a, a1, b, c, d, \kappa, \lambda, m, n, pterm, p_0, p_1, p_2, p_3, p_4, q_0, q_1, q_2, \frac{1}{n^{8/3}}, \frac{1}{n^{7/3}}, \right. \tag{2.7.2.2}$$

$$\frac{1}{n^{7/6}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{2/3}}, \frac{1}{n^{1/3}}, \frac{1}{(n-3)^{1/3}}, \frac{1}{(n-2)^{7/6}},$$

$$\frac{1}{(n-2)^{2/3}}, \frac{1}{(n-2)^{1/3}}, \frac{1}{(n-1)^{7/6}}, \frac{1}{(n-1)^{2/3}}, \frac{1}{(n-1)^{1/3}} \}$$

The error terms are (to check, look at posFc)

```
> simplify(O((2^(1/3)*(m+1)/n^(1/3)-2^(1/3)*m/n^(1/3))
  ^ordAiUp));
  simplify(O((2^(1/3)*(m)/(n-1)^(1/3)-2^(1/3)*m/n^(1/3))
  ^ordAiUp));
  simplify(O((2^(1/3)*(m+2)/(n-1)^(1/3)-2^(1/3)*m/n^(1/3))
  ^ordAiUp));
```

$$O\left(\frac{64\,2^{1/3}}{n^{19/3}}\right)$$

$$O\left(\frac{64\,2^{1/3}\,m^{19}\left(n^{1/3}-(n-1)^{1/3}\right)^{19}}{(n-1)^{19/3}\,n^{19/3}}\right)$$

$$O\left(\frac{64\left(-m\,(n-1)^{1/3}+n^{1/3}\,(m+2)\right)^{19}\,2^{1/3}}{n^{19/3}\,(n-1)^{19/3}}\right) \quad \textbf{(2.7.2.3)}$$

remove error terms

```
> posFd := convert(posF,polynom):
```

Next we fit the display size of the plots in the n (Nord) and m (Mord) direction.
(Note that everything up to ordAi is computed, but possibly not shown)

```
> Nord := -ordAiUp/3;
  Mord := floor(ordAiUp/3)+1;
  myview := view=[0..Mord,Nord..0]:
```

$$Nord := -\frac{19}{3}$$

$$Mord := 7 \quad \textbf{(2.7.2.4)}$$

Expand again with respect to n,
these are then our unknowns

```
> posFe := series(posFd,n=infinity,ceil(-Nord)+1):indets
  (%);
  posFf := convert(%%,polynom):
```

$$\Big\{a, a1, b, c, d, \kappa, \lambda, m, n, pterm, p_0, p_1, p_2, p_3, p_4, q_0, q_1, q_2, \left(\frac{1}{n}\right)^{1/3}, \left(\frac{1}{n}\right)^{2/3}, \quad \textbf{(2.7.2.5)}$$
$$\left(\frac{1}{n}\right)^{3/2}, \left(\frac{1}{n}\right)^{4/3}, \left(\frac{1}{n}\right)^{5/2}, \left(\frac{1}{n}\right)^{5/3}, \left(\frac{1}{n}\right)^{7/2}, \left(\frac{1}{n}\right)^{7/3}, \left(\frac{1}{n}\right)^{7/6},$$
$$\left(\frac{1}{n}\right)^{8/3}, \left(\frac{1}{n}\right)^{9/2}, \left(\frac{1}{n}\right)^{10/3}, \left(\frac{1}{n}\right)^{11/2}, \left(\frac{1}{n}\right)^{11/3}, \left(\frac{1}{n}\right)^{11/6},$$
$$\left(\frac{1}{n}\right)^{13/2}, \left(\frac{1}{n}\right)^{13/3}, \left(\frac{1}{n}\right)^{13/6}, \left(\frac{1}{n}\right)^{14/3}, \left(\frac{1}{n}\right)^{16/3}, \left(\frac{1}{n}\right)^{17/3},$$
$$\left(\frac{1}{n}\right)^{17/6}, \left(\frac{1}{n}\right)^{19/3}, \left(\frac{1}{n}\right)^{19/6}, \left(\frac{1}{n}\right)^{20/3}, \left(\frac{1}{n}\right)^{23/6}, \left(\frac{1}{n}\right)^{25/6},$$
$$\left(\frac{1}{n}\right)^{29/6}, \left(\frac{1}{n}\right)^{31/6}, \left(\frac{1}{n}\right)^{35/6}, \left(\frac{1}{n}\right)^{37/6}, \left(\frac{1}{n}\right)^{41/6}, O\left(\frac{1}{n^7}\right)\Big\}$$

The mynewt function computes the Newton polygon of posFf

```
> newt1 := mynewt(posFf,m,n):
```

First Newton polygon, where no unknowns have been fixed

```
> P1 := pointplot(newt1,myoptionsUp,color=blue):
  display(P1,myview);
```



Here, we want to kill the element (0,0)

```
> getel(posFf,0,0);
```

$$\kappa a^3 - 2\,\kappa a^2 \quad (2.7.2.6)$$

Set a=2

```
> posFfa := expand(simplify(subs(a=2,posFf))) assuming
  n::posint,m::posint:
> newta := mynewt(posFfa,m,n):
```

All blue points have been eliminated, and only the red ones remain

```
> P1a := pointplot(newta,myoptionsUp,color=red):
  display(P1,P1a,myview);
```



b=0 is forced due to the term n^{-1/3)

```
> getel(posFfa,0,-1/3);
```

$$\frac{4\,\kappa\,b}{n^{1/3}} \qquad \textbf{(2.7.2.7)}$$

set a=2, b=0

```
> posFfab := expand(simplify(subs(b=0,posFfa))) assuming
  n::posint,m::posint:
> newtab := mynewt(posFfab,m,n):
```

Now only the green points remain.

```
> P1ab := pointplot(newtab,myoptionsUp,color=green):
  display(P1,P1a,P1ab,myview);
```



at this point we find our choice for c, which we heuristically computed already before in Section 3

```
> csubs;
  factor(getel(posFfab,0,-2/3));#subs(csubs,%);
  factor(getel(posFfab,1,-5/3));#subs(csubs,%);
  factor(getel(posFfab,2,-5/3));#subs(csubs,%);
```

$$c = a1\, 2^{2/3}$$

$$-\frac{4\,\kappa\,(a1\, 2^{2/3} - c)}{n^{2/3}}$$

$$-\frac{4\,\kappa\, m\left(a1\, 2^{2/3} q_1 + 4\, a1\, 2^{2/3} q_2 + 3\, a1\, 2^{2/3} - c\, q_1 - c\right)}{n^{5/3}}$$

$$-\frac{4\,\kappa\, m^2 q_2\,(a1\, 2^{2/3} - c)}{n^{5/3}} \qquad \textbf{(2.7.2.8)}$$

set a=2, b=0, c=a1*2^(2/3)

```
> posFfabc := expand(simplify(subs(csubs,posFfab)))
  assuming n::posint,m::posint:
> newtabc := mynewt(posFfabc,m,n):
> P1abc := pointplot(newtabc,myoptionsUp,color=yellow):
  display(P1,P1a,P1ab,P1abc,myview);
```



Here we get q[2] and pterm

```
> factor(getel(posFfabc,0,-1));
  factor(getel(posFfabc,1,-4/3));
  solve({%%,%},{q[2],pterm});
```

$$\frac{2\,\kappa\,(2\,pterm - 4\,q_2 - 7)}{n}$$

$$-\frac{16\,\lambda\,2^{1/3}\,m\,(2 + 3\,q_2)}{3\,n^{4/3}}$$

$$\left\{pterm = \frac{13}{6},\ q_2 = -\frac{2}{3}\right\} \qquad \textbf{(2.7.2.9)}$$

set a=2, b=0,c,pterm=13/6,q[2]=-2/3

```
> posFfabcd := expand(simplify(subs(pterm=13/6,q[2]=-2/3,
  posFfabc))) assuming n::posint,m::posint:
> newtabcd := mynewt(posFfabcd,m,n):
```

only the brown points remain
Now all points are strictly below n^{-1}

```
> P1abcd := pointplot(newtabcd,myoptionsUp,color=brown):
  display(P1,P1a,P1ab,P1abc,P1abcd,myview);
```



Here are the dominating corners and we see that we have to choose d=1 to have a positive term;
note that we will see that the second term should be negative, as lambda=Ai' is negative for large m,
hence here we will choose a p[4]>2/9 (see below in the decomposition into lambda and kappa contributions)

```
> getel(posFfabcd,0,-7/6);
  factor(getel(posFfabcd,3,-14/6));
  (14/6-7/6)/3;   #slope
```

$$\frac{4\,\kappa\,d}{n^{7/6}}$$

$$-\frac{32\,2^{1/3}\,\lambda\,m^3\,(-2+9\,p_4)}{9\,n^{7/3}}$$

$$\frac{7}{18} \tag{2.7.2.10}$$

and continuing

```
> getel(posFfabcd,5,-20/6);
  (20/6-14/6)/2;    #slope
```

$$-\frac{32\,2^{1/3}\,\lambda\,m^5\,p_4}{3\,n^{10/3}}$$

$$\frac{1}{2} \tag{2.7.2.11}$$

set a=2, b=0,c,pterm=13/6,q[2]=-2/3 and d=1

```
> posFfabcde := expand(simplify(subs(d=1,posFfabcd)))
  assuming n::posint,m::posint:
> newtabcde := mynewt(posFfabcde,m,n):
```

This is the final result, where only the solid diamonds are non-zero

```
> P1abcde := pointplot(newtabcde,myoptionsUp,symbol=
  soliddiamond,color=gray):
  display(P1,P1a,P1ab,P1abc,P1abcd,P1abcde,myview);
```



Plot the boundary and the slopes of the Newton polygon;
Note that we have already proved that there are now points above the blue dotted line

```
> LegendSize := size=[600,450]:
  P1dom1 := plot(-7/6-(7/18)*m,m=0..3,color=black,legend=
  [typeset("Slope: ", -7/18)],legendstyle=[location=right]
  ):
  P1dom2 := plot(-5/6-(1/2)*m,m=3..5,color=red,legend=
  [typeset("Slope: ", -1/2)],legendstyle=[location=right])
  :
  P1dom2b := plot(0-(2/3)*m,m=5..6,color=green,legend=
  [typeset("Slope: ", -2/3)],legendstyle=[location=right])
  :
  P1dom3a := plot(2-m,m=0..7,color=blue,linestyle=dot,
  legend=[typeset("Slope: ", -1)],legendstyle=[location=
  right]):
  P1all := display(P1,P1a,P1ab,P1abc,P1abcd,P1abcde,
  P1dom1,P1dom2,P1dom2b,P1dom3a,myview,LegendSize):
```

```
for i from 1 to 10 do
    P1dom3[i] := plot([[2+i/2,0],[2+i/2,-i/2]],color=
blue,linestyle=dot):
end:
display(P1all,seq(P1dom3[i],i=1..10));
```



This is the choice for SF

```
> SF(n);
  subs(a=2,b=0,csubs,pterm=13/6,d=1,%);
```

$$a+\frac{b}{n^{1/3}}+\frac{c}{n^{2/3}}+\frac{pterm}{n}+\frac{d}{n^{7/6}}$$

$$2+\frac{a1\, 2^{2/3}}{n^{2/3}}+\frac{13}{6\, n}+\frac{1}{n^{7/6}} \qquad \textbf{(2.7.2.12)}$$

recall

```
> kaplam;
```

$$\mathrm{AiryAi}\left(\mathrm{AiryAiZeros}(1)+\frac{2^{1/3}\, m}{n^{1/3}}\right)=\kappa, \mathrm{AiryAi}\left(1, \mathrm{AiryAiZeros}(1)+\frac{2^{1/3}\, m}{n^{1/3}}\right)=\lambda \qquad \textbf{(2.7.2.13)}$$

Look at the corners
Here we see the necessary choice: p[4]>2/9 (note that for m>n^(1/3) lambda is negative)

```
> getel(posFfabcde,0,-7/6);
  simplify(getel(posFfabcde,3,-14/6));
  getel(posFfabcde,5,-20/6);
  getel(posFfabcde,6,-4);
  getel(posFfabcde,7,-5);
```

$$\frac{4\,\kappa}{n^{7/6}}$$

$$-\frac{32\,2^{1/3}\,\lambda\,m^3\,(-2+9\,p_4)}{9\,n^{7/3}}$$

$$-\frac{32\,2^{1/3}\,\lambda\,m^5\,p_4}{3\,n^{10/3}}$$

$$-\frac{68\,\kappa\,m^6\,p_4}{3\,n^4}$$

$$-\frac{124\,\kappa\,m^7\,p_4}{45\,n^5} \qquad \textbf{(2.7.2.14)}$$

Now split the black dots into the contributions from Ai and Ai'

```
> indets(posFfabcd);
```

$$\left\{a1, d, \kappa, \lambda, m, n, p_0, p_1, p_2, p_3, p_4, q_0, q_1, \frac{1}{n^{41/6}}, \frac{1}{n^{37/6}}, \frac{1}{n^{35/6}}, \frac{1}{n^{31/6}}, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{20/3}}, \frac{1}{n^{19/3}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/3}}, \frac{1}{n^{17/6}}, \frac{1}{n^{16/3}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/2}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/2}}, \frac{1}{n^{11/3}}, \frac{1}{n^{11/6}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{7/6}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{3/2}}\right\} \qquad \textbf{(2.7.2.15)}$$

Sanity check that there are no other contributions

```
> subs(kappa=0,lambda=0,posFfabcd);
```

$$0 \qquad \textbf{(2.7.2.16)}$$

Extract the coefficients of kappa=Ai and lambda=Ai'
and treat then seperately

```
> posFk := coeff(posFfabcd,kappa):indets(%);
  posFl := coeff(posFfabcd,lambda):indets(%);
```

$$\left\{a1, d, m, n, p_0, p_1, p_2, p_3, p_4, q_0, q_1, \frac{1}{n^{41/6}}, \frac{1}{n^{37/6}}, \frac{1}{n^{35/6}}, \frac{1}{n^{31/6}}, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{20/3}}, \frac{1}{n^{19/3}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/3}}, \frac{1}{n^{17/6}}, \frac{1}{n^{16/3}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/2}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/2}}, \frac{1}{n^{11/3}}, \frac{1}{n^{11/6}}, \frac{1}{n^{10/3}},\right.$$

$\frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{7/6}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}\Big\}$

$$\Big\{a1, d, m, n, p_0, p_1, p_2, p_3, p_4, q_0, q_1, \frac{1}{n^{41/6}}, \frac{1}{n^{37/6}}, \frac{1}{n^{35/6}}, \frac{1}{n^{31/6}}, \frac{1}{n^{29/6}}, \frac{1}{n^{25/6}}, \frac{1}{n^{23/6}}, \frac{1}{n^{20/3}}, \frac{1}{n^{19/3}}, \frac{1}{n^{19/6}}, \frac{1}{n^{17/3}}, \frac{1}{n^{17/6}}, \frac{1}{n^{16/3}}, \frac{1}{n^{14/3}}, \frac{1}{n^{13/2}}, \frac{1}{n^{13/3}}, \frac{1}{n^{13/6}}, \frac{1}{n^{11/2}}, \frac{1}{n^{11/3}}, \frac{1}{n^{10/3}}, \frac{1}{n^{9/2}}, \frac{1}{n^{8/3}}, \frac{1}{n^{7/2}}, \frac{1}{n^{7/3}}, \frac{1}{n^{5/2}}, \frac{1}{n^{5/3}}, \frac{1}{n^{4/3}}, \frac{1}{n^{3/2}}\Big\} \quad \textbf{(2.7.2.17)}$$

We color the non-zero nodes of the lastNewton polygon into
red squared .... coefficients of kappa=Ai
blue diamonds ...... coefficients of lambda=Ai'

```
> newt4a := mynewt(posFk,m,n):
  newt4b := mynewt(posFl,m,n):
> P4a := pointplot(newt4a,labels=["m deg", "n deg"],
  symbolsize=15, symbol=box,color=red):
  P4b := pointplot(newt4b,labels=["m deg", "n deg"],
  symbolsize=15, symbol=diamond, color=blue):
  P1dom3s := plot(1-m,m=4..5,color=black):
  display(P4a,P4b,P1dom1,P1dom2,P1dom2b,P1dom3a,myview,
  myoptionsUp,LegendSize);
```



red extremes of Newton polygon

```
> mnmaxR := getMaxNewt(Mord,newt4a):
  seq([i,mnmaxR[i]],i=0..Mord);
```

$$\left[0, -\frac{7}{6}\right], \left[1, -\frac{5}{3}\right], [2, -2], \left[3, -\frac{8}{3}\right], [4, -3], \left[5, -\frac{11}{3}\right], [6, -4], [7, -5] \quad \textbf{(2.7.2.18)}$$

These are the specific values at these points;
we see that we still have some degree in freedom: d and p[4]

```
> for i from 0 to Mord do
     i,factor(getel(posFk,i,mnmaxR[i]));
  end;
```

$$0, \frac{4\,d}{n^{7/6}}$$

$$1, \frac{8\,a1\,2^{2/3}\,m}{3\,n^{5/3}}$$

$$2, -\frac{4\,m^2\,(-41 + 108\,p_4)}{9\,n^2}$$

$$3, -\frac{16\, 2^{2/3}\, a1\, m^3\, (6\, p_4 - 1)}{3\, n^{8/3}}$$
$$4, -\frac{8\, m^4\, (-17 + 132\, p_4)}{9\, n^3}$$
$$5, -\frac{8\, 2^{2/3}\, a1\, m^5\, p_4}{n^{11/3}}$$
$$6, -\frac{68\, m^6\, p_4}{3\, n^4}$$
$$7, -\frac{124\, m^7\, p_4}{45\, n^5} \tag{2.7.2.19}$$

These are the slopes of the convex hull where the corners are given by the second sequence;
hence, in order to be positive when the slope > -1, we need to choose d>0, e.g. d=1;
note that p[4] is not important here, as the slope first slope -5/12 is less then -7/18;
and in the later regimes it will be dominated by the blue points.

```
> ls,li := getslopes(mnmaxR,Mord);
```

$$ls, li := \left[-\frac{5}{12}, -\frac{1}{2}, -1\right], [0, 2, 6, 7] \tag{2.7.2.20}$$

```
> colors := [green,black,brown,blue,olive,red]:
  styles := [spacedot,solid, dash, dot, dashdot, longdash,
  spacedash]:
```

Draw the conveux hull in red

```
> for i from 1 to nops(ls) do
     ls[i];li[i];
     tt[i] := plot((mnmaxR[li[i]]-ls[i]*li[i])+ls[i]*m,m=
  li[i]..li[i+1],color=colors[i mod nops(colors)+1],
  linestyle=styles[i mod nops(styles)+1],legend=[typeset
  ("Slope: ", ls[i])],legendstyle=[location=right]):
  end:
  Pconvred := seq(tt[i],i=1..nops(ls)):
  #display(%);
> display(Pconvred,P4a,myview,myoptionsUp,LegendSize);
```



We continue with the blue diamonds, i.e., the coefficients of Ai'

```
> mnmaxB := getMaxNewt(Mord,newt4b):
  seq([i,mnmaxB[i]],i=0..Mord);
```

$$\left[0, -\frac{4}{3}\right], [1, -2], \left[2, -\frac{7}{3}\right], \left[3, -\frac{7}{3}\right], \left[4, -\frac{10}{3}\right], \left[5, -\frac{10}{3}\right], \left[6, -\frac{13}{3}\right], \left[7, -\frac{16}{3}\right] \quad \textbf{(2.7.2.21)}$$

```
> for i from 0 to Mord do
     i,factor(getel(posFl,i,mnmaxB[i]));
  end;
```

$$0, -\frac{2\, 2^{1/3} (4 q_1 - 1)}{n^{4/3}}$$

$$1, \frac{32\, a1\, m}{9 n^2}$$

$$2, -\frac{8\, 2^{1/3} m^2 (12 q_1 + 54 p_4 + 27 p_3 - 17)}{9 n^{7/3}}$$

$$3,\ -\frac{32\,2^{1/3}\,m^3\,(-2+9\,p_4)}{9\,n^{7/3}}$$

$$4,\ -\frac{2\,2^{1/3}\,m^4\,(-20+48\,p_3+279\,p_4)}{9\,n^{10/3}}$$

$$5,\ -\frac{32\,2^{1/3}\,m^5\,p_4}{3\,n^{10/3}}$$

$$6,\ -\frac{20\,2^{1/3}\,m^6\,p_4}{3\,n^{13/3}}$$

$$7,\ -\frac{356\,2^{1/3}\,m^7\,p_4}{45\,n^{16/3}} \qquad \textbf{(2.7.2.22)}$$

Again we derive the slopes of the convex hull and its corners;
note that if we choose q[1]=1/4 we eliminate the first term and decrease the slope.
This will be useful in the proof of Lemma 5.3, in the same way as it was used in the proof of Lemma 4.4.

```
> ls,li := getslopes(mnmaxB,Mord);
```

$$ls,\ li := \left[-\frac{1}{3},\ -\frac{1}{2},\ -1\right],\ [0, 3, 5, 7] \qquad \textbf{(2.7.2.23)}$$

```
> colors := [brown,olive,purple,blue,olive,red,black]:
  styles := [spacedot,solid, dash, dot, dashdot, longdash,
  spacedash]:
```

Draw the conveux hull in blue which still includes the term of order Theta(n^(-4/3))

```
> for i from 1 to nops(ls) do
     ls[i];li[i];
     tt[i] := plot((mnmaxB[li[i]]-ls[i]*li[i])+ls[i]*m,m=
  li[i]..li[i+1],color=colors[i mod nops(colors)+1],
  linestyle=styles[i mod nops(styles)+1],legend=[typeset
  ("Slope: ", ls[i])],legendstyle=[location=right]):
  end:
  Pconvblue := seq(tt[i],i=1..nops(ls)):
  #display(%);
> display(Pconvblue,P4b,myview,myoptionsUp,LegendSize);
```



We kill this term by setting q[1] = 1/4 and recompute the Newton polygons.
In the next picture the left-top blue point disappeared.
(note that the coefficients of lower order terms of posFk change as well, which is why we recompute them; however the picture will not change as the convex hull is not influenced by this change.)

```
> posFk2 := coeff(subs(q[1]=1/4,p[3]=0,p[2]=0,p[1]=0,p[0]=
  0,q[0]=0,posFfabcde),kappa):indets(%):
  newt4a2 := mynewt(posFk2,m,n):
  P4a2 := pointplot(newt4a2,labels=["m deg", "n deg"],
  symbolsize=15, symbol=box, color=red):
  posFl2 := coeff(subs(q[1]=1/4,p[3]=0,p[2]=0,p[1]=0,p[0]=
  0,q[0]=0,posFfabcde),lambda):indets(%):
  newt4b2 := mynewt(posFl2,m,n):
  P4b2 := pointplot(newt4b2,labels=["m deg", "n deg"],
  symbolsize=15, symbol=diamond, color=blue):
  display(P4a2,P4b2,P1dom1,P1dom2,P1dom2b,P1dom3a,myview,
  myoptionsUp,LegendSize);
```



Recompute blue with q[1]=1/4

```
> mnmaxB := getMaxNewt(Mord,newt4b2):
  seq([i,mnmaxB[i]],i=0..Mord);
```

$$\left[0, -\frac{3}{2}\right], [1, -2], \left[2, -\frac{7}{3}\right], \left[3, -\frac{7}{3}\right], \left[4, -\frac{10}{3}\right], \left[5, -\frac{10}{3}\right], \left[6, -\frac{13}{3}\right], \left[7, -\frac{16}{3}\right] \quad \textbf{(2.7.2.24)}$$

```
> for i from 0 to Mord do
     i,factor(getel(posFl2,i,mnmaxB[i]));
  end;
```

$$0, \frac{4\, 2^{1/3}}{n^{3/2}}$$

$$1, \frac{32\, a1\, m}{9\, n^2}$$

$$2, -\frac{16\, 2^{1/3} m^2 \left(-7 + 27\, p_4\right)}{9\, n^{7/3}}$$

$$3,\ -\frac{32\ 2^{1/3}\ m^3\ (-2+9\ p_4)}{9\ n^{7/3}}$$

$$4,\ -\frac{2\ 2^{1/3}\ m^4\ (279\ p_4-20)}{9\ n^{10/3}}$$

$$5,\ -\frac{32\ 2^{1/3}\ m^5\ p_4}{3\ n^{10/3}}$$

$$6,\ -\frac{20\ 2^{1/3}\ m^6\ p_4}{3\ n^{13/3}}$$

$$7,\ -\frac{356\ 2^{1/3}\ m^7\ p_4}{45\ n^{16/3}} \quad \textbf{(2.7.2.25)}$$

And we get new slopes, yet at the same m powers given in the second sequence;
here we need that the term m^3 is negative as it will dominate in the regime when Ai' is negative,
therefore we have to choose p[4]>2/9 (note that this also makes m^5 and m^7 multiplied by Ai' positive)

```
> ls,li := getslopes(mnmaxB,Mord);
```

$$ls, li := \left[-\frac{5}{18}, -\frac{1}{2}, -1\right], [0, 3, 5, 7] \quad \textbf{(2.7.2.26)}$$

Draw the new conveux hull in blue

```
> for i from 1 to nops(ls) do
    ls[i];li[i];
    tt[i] := plot((mnmaxB[li[i]]-ls[i]*li[i])+ls[i]*m,m=
  li[i]..li[i+1],color=colors[i mod nops(colors)+1],
  linestyle=styles[i mod nops(styles)+1],legend=[typeset
  ("Slope: ", ls[i])],legendstyle=[location=right]):
  end:
  Pconvblue := seq(tt[i],i=1..nops(ls)):
  #display(%);
```

Plot the difference to before:
Only the solid circle on the top-left disappeared.

```
> P4bdiffshort := pointplot([0,-4/3],labels=["m deg", "n
  deg"], symbolsize=15, symbol=solidcircle, color=blue):
  display(Pconvblue,P4bdiffshort,P4b2,myview,myoptionsUp,
  LegendSize);
```



Lemma 4.5

We want to prove
for integers 0 <= j < k <= l <= 2*n,
k-j even, that

```
> p(l,j,2*n)/(j+1) >= p(l,k,2*n)/(k+1)
```

$$\frac{p(l, k, 2\,n)}{k+1} \le \frac{p(l, j, 2\,n)}{j+1} \quad \textbf{(3.1)}$$

It suffices to prove that the following is non-negative

```
> LLp := p(l,m-1,2*n)/m-p(l,m+1,2*n)/(m+2);
```

(3.2)

$$LLp := \frac{p(l, m-1, 2n)}{m} - \frac{p(l, m+1, 2n)}{m+2} \quad (3.2)$$

this is the recurrence holding for m non-negative

```
> prec := p(l,m,2*n) = p(l+1,m-1,2*n) + (l-m+2)/(l+m+2)*p(l+1,
  m+1,2*n);
```

$$prec := p(l, m, 2n) = p(l+1, m-1, 2n) + \frac{(l-m+2)\, p(l+1, m+1, 2n)}{l+m+2} \quad (3.3)$$

These relations are used in the induction step

```
> pup := factor(subs(m=m+1,isolate(LLp,p(l,m-1,2*n))));
  pdown := factor(subs(m=m-1,isolate(LLp,p(l,m+1,2*n))));
```

$$pup := p(l, m, 2n) = \frac{p(l, m+2, 2n)\, (m+1)}{m+3}$$

$$pdown := p(l, m, 2n) = \frac{p(l, m-2, 2n)\, (m+1)}{m-1} \quad (3.4)$$

Now, we perform an induction on l, with l=2*n being the base case.
We start by applying the recurrence to LLp.

```
> pe1 := collect(subs(subs(m=m-1,prec),subs(m=m+1,prec),LLp),
  p,factor);
```

$$pe1 := \frac{2\, (-m^2 + l + 3)\, p(l+1, m, 2n)}{(l+m+1)\, m\, (m+2)} + \frac{p(l+1, m-2, 2n)}{m} - \frac{(l-m+1)\, p(l+1, m+2, 2n)}{(l+m+3)\, (m+2)} \quad (3.5)$$

Now all l's are larger than m, so we can use the inductive hypothesis

Here we use the induction hypothesis and get the rational function which is clearly positive for l>=m>=1

```
> pe2 := collect(subs(
      subs(l=l+1,m=m-2,pup),
      subs(l=l+1,m=m+2,pdown),
  pe1), p,factor);
```

$$pe2 := \frac{4\, (l\, m + l + m + 3)\, p(l+1, m, 2n)}{(l+m+3)\, (l+m+1)\, m\, (m+2)} \quad (3.6)$$

Lemma 5.4

We want to prove
for integers 0 <= j < k <= l <= 2*n,
k-j even,
and n large (here >= 10) that

```
> q(l,j,2*n)/(j+1) >= q(l,k,2*n)/(k+1)
```

$$\frac{q(l, k, 2n)}{k+1} \le \frac{q(l, j, 2n)}{j+1} \quad (4.1)$$

It suffices to prove that the following is non-negative

```
> LLq := q(l,m-1,2*n)/m-q(l,m+1,2*n)/(m+2);
```

$$LLq := \frac{q(l, m-1, 2n)}{m} - \frac{q(l, m+1, 2n)}{m+2} \quad (4.2)$$

This is the recurrence holding for m non-negative

```
> qrec := q(l,m,2*n) = (l-m)/(l-m+2)*q(l+1,m-1,2*n)
                      + (l-m+2)/(l+m+2)*q(l+1,m+1,2*n)
                      + 2/(l-m+4)*q(l+2,m-2,2*n)
                      + 2/(l+m+2)*q(l+3,m-1,2*n)
                      + 4/(l+m+4)/(l+m+2)*q(l+3,m+1,2*n);
```

$$qrec := q(l, m, 2\,n) = \frac{(l-m)\,q(l+1, m-1, 2\,n)}{l-m+2} + \frac{(l-m+2)\,q(l+1, m+1, 2\,n)}{l+m+2} + \frac{2\,q(l+2, m-2, 2\,n)}{l-m+4} + \frac{2\,q(l+3, m-1, 2\,n)}{l+m+2} + \frac{4\,q(l+3, m+1, 2\,n)}{(l+m+4)\,(l+m+2)} \qquad (4.3)$$

These relations are used in the induction step

```
> qup := subs(m=m+1,isolate(LLq,q(l,m-1,2*n)));
  qdown := subs(m=m-1,isolate(LLq,q(l,m+1,2*n)));
```

$$qup := q(l, m, 2\,n) = \frac{q(l, m+2, 2\,n)\,(m+1)}{m+3}$$

$$qdown := q(l, m, 2\,n) = \frac{q(l, m-2, 2\,n)\,(m+1)}{m-1} \qquad (4.4)$$

Now, we perform an induction on l, with l=2*n being the base case.
We start by applying the recurrence to LLq.

```
> qe1 := collect(subs(subs(m=m-1,qrec),subs(m=m+1,qrec),LLq),
  q,factor);
```

$$qe1 := \frac{2\,(-l\,m^2+m^3+l^2+4\,l-2\,m+3)\,q(l+1, m, 2\,n)}{(l+m+1)\,m\,(l-m+1)\,(m+2)} + \frac{(l-m+1)\,q(l+1, m-2, 2\,n)}{(l-m+3)\,m} - \frac{(l-m+1)\,q(l+1, m+2, 2\,n)}{(l+m+3)\,(m+2)} + \frac{2\,q(l+2, m-3, 2\,n)}{(l-m+5)\,m} - \frac{2\,q(l+2, m-1, 2\,n)}{(l-m+3)\,(m+2)} - \frac{2\,(l\,m+m^2-m-4)\,q(l+3, m, 2\,n)}{(l+m+3)\,(l+m+1)\,m\,(m+2)} + \frac{2\,q(l+3, m-2, 2\,n)}{(l+m+1)\,m} - \frac{4\,q(l+3, m+2, 2\,n)}{(l+m+5)\,(l+m+3)\,(m+2)} \qquad (4.5)$$

Now all l's are larger than m, so we can use the inductive hypothesis

Here we use the induction hypothesis and get the rational functions R1, R2, R3
(Note that if m=2, then the term q(l+2,m-3,2*n) is actually 0; but our qup-shift makes it 0)

```
> qe2 := collect(subs(
      subs(l=l+1,m=m-2,qup),
      subs(l=l+1,m=m+2,qdown),
      subs(l=l+2,m=m-3,qup),
      subs(l=l+3,m=m+2,qdown),
      subs(l=l+3,m=m,qdown),
  qe1), q,factor);
```

$$qe2 := \big(4\,(l+2)\,(l^2\,m^2-2\,l\,m^3+m^4+2\,l^2\,m+2\,m^3+2\,l^2+8\,l\,m-3\,m^2+8\,l+2\,m+6)\,q(l+1, m, 2\,n)\big)\big/\big((l+m+1)\,m\,(l-m+1)\,(m+2)\,(l-m+3)\,(m+1)\,(l+m+3)\big) - \frac{4\,(m^2+2\,l-2\,m+6)\,q(l+2, m-1, 2\,n)}{(l-m+5)\,m^2\,(l-m+3)\,(m+2)} - \frac{4\,(-l\,m^2-m^3+l^2+l\,m-9\,m^2+6\,l-11\,m+5)\,q(l+3, m-2, 2\,n)}{(l+m+3)\,(l+m+1)\,m\,(m+2)\,(m-1)\,(l+m+5)} \qquad (4.6)$$

Next, we check if coefficient of q(l+1,m,2n) is non-negative
the denominator is positive, so we only deal with the numerator

```
> coeff(qe2,q(l+1,m,2*n));
  numer(%);
  simplify(subs(m=a,l=a+b,%));
```

$$\frac{4\,(l+2)\,(l^2 m^2 - 2\,l\,m^3 + m^4 + 2\,l^2 m + 2\,m^3 + 2\,l^2 + 8\,l\,m - 3\,m^2 + 8\,l + 2\,m + 6)}{(l+m+1)\,m\,(l-m+1)\,(m+2)\,(l-m+3)\,(m+1)\,(l+m+3)}$$

$$4\,(l+2)\,(l^2 m^2 - 2\,l\,m^3 + m^4 + 2\,l^2 m + 2\,m^3 + 2\,l^2 + 8\,l\,m - 3\,m^2 + 8\,l + 2\,m + 6)$$

$$16\,(a+b+2)\left(a^3 + \left(\frac{1}{4}\,b^2 + b + \frac{7}{4}\right)a^2 + \left(\frac{1}{2}\,b^2 + 3\,b + \frac{5}{2}\right)a + \frac{b^2}{2} + 2\,b + \frac{3}{2}\right) \qquad \textbf{(4.7)}$$

Finally, we use the following simple bounds, arising by taking just the first term on the right-hand side of qrec;
Note that we DO NOT have equality here but a >= sign;
however, for the subsitution process we use it like that.

```
> qrecUpBound := q(l,m,2*n) = (l-m)/(l-m+2)*q(l+1,m-1,2*n);
```

$$qrecUpBound := q(l, m, 2\,n) = \frac{(l-m)\,q(l+1, m-1, 2\,n)}{l-m+2} \qquad \textbf{(4.8)}$$

We will use this bound

```
> subs(l=l+1,qrecUpBound);
```

$$q(l+1, m, 2\,n) = \frac{(l-m+1)\,q(l+2, m-1, 2\,n)}{l-m+3} \qquad \textbf{(4.9)}$$

Now all are the same

```
> qe3 := collect(subs(
      subs(l=l+1,qrecUpBound),
  qe2), q,factor);
```

$$qe3 := \Big(4\,(l^4 m^3 - 3\,l^3 m^4 + 3\,l^2 m^5 - l\,m^6 + 2\,l^4 m^2 + 4\,l^3 m^3 - 15\,l^2 m^4 + 10\,l\,m^5 - m^6 + 19\,l^3 m^2 - 9\,l^2 m^3 - 8\,l\,m^4 + 6\,m^5 - 2\,l^4 + 2\,l^3 m + 59\,l^2 m^2 - 40\,l\,m^3 + 20\,m^4 - 20\,l^3 + 6\,l^2 m + 53\,l\,m^2 - 24\,m^3 - 72\,l^2 - 10\,l\,m - 13\,m^2 - 108\,l - 30\,m - 54)\,q(l+2, m-1, 2\,n)\Big) \Big/ \big((l+m+1)\,m^2\,(m+2)\,(l-m+3)^2\,(m+1)\,(l+m+3)\,(l-m+5)\big) - \frac{4\,(-l\,m^2 - m^3 + l^2 + l\,m - 9\,m^2 + 6\,l - 11\,m + 5)\,q(l+3, m-2, 2\,n)}{(l+m+3)\,(l+m+1)\,m\,(m+2)\,(m-1)\,(l+m+5)} \qquad \textbf{(4.10)}$$

Then, we check if coefficient of q(l+2,m-1,2n) is non-negative
the denominator is positive, so we only deal with the numerator

```
> coeff(qe3,q(l+2,m-1,2*n));
  numer(%);
  simplify(subs(m=a+2,l=a+b+2,%));
```

$$\Big(4\,(l^4 m^3 - 3\,l^3 m^4 + 3\,l^2 m^5 - l\,m^6 + 2\,l^4 m^2 + 4\,l^3 m^3 - 15\,l^2 m^4 + 10\,l\,m^5 - m^6 + 19\,l^3 m^2 - 9\,l^2 m^3 - 8\,l\,m^4 + 6\,m^5 - 2\,l^4 + 2\,l^3 m + 59\,l^2 m^2 - 40\,l\,m^3 + 20\,m^4 - 20\,l^3 + 6\,l^2 m + 53\,l\,m^2 - 24\,m^3 - 72\,l^2 - 10\,l\,m - 13\,m^2 - 108\,l - 30\,m - 54)\Big) \Big/ \big((l+m+1)\,m^2\,(m+2)\,(l-m+3)^2\,(m+1)\,(l+m+3)\,(l-m+5)\big)$$

$$4\,l^4 m^3 - 12\,l^3 m^4 + 12\,l^2 m^5 - 4\,l\,m^6 + 8\,l^4 m^2 + 16\,l^3 m^3 - 60\,l^2 m^4 + 40\,l\,m^5 - 4\,m^6 + 76\,l^3 m^2 - 36\,l^2 m^3 - 32\,l\,m^4 + 24\,m^5 - 8\,l^4 + 8\,l^3 m + 236\,l^2 m^2 - 160\,l\,m^3 + 80\,m^4 - 80\,l^3 + 24\,l^2 m + 212\,l\,m^2 - 96\,m^3 - 288\,l^2 - 40\,l\,m - 52\,m^2 - 432\,l$$

$$-120\,m-216$$

$$32\,a^5+\left(4\,b^3+36\,b^2+124\,b+476\right)a^4+\left(4\,b^4+80\,b^3+480\,b^2+1296\,b+2588\right)a^3+\left(32\,b^4+460\,b^3+2228\,b^2+4820\,b+6284\right)a^2+80\left(b^2+\frac{23}{10}\,b+\frac{31}{10}\right)(b+5)^2\,a+56\,(b+1)^2\left(b+\frac{29}{7}\right)(b+5) \qquad \textbf{(4.11)}$$

Next, we check if coefficient of q(l+1,m,2n) is non-negative
the denominator is positive, so we only deal with the numerator

```
> coeff(qe2,q(l+1,m,2*n));
  numer(%);
  simplify(subs(m=a,l=a+b,%));
```

$$\frac{4\,(l+2)\left(l^2\,m^2-2\,l\,m^3+m^4+2\,l^2\,m+2\,m^3+2\,l^2+8\,l\,m-3\,m^2+8\,l+2\,m+6\right)}{(l+m+1)\,m\,(l-m+1)\,(m+2)\,(l-m+3)\,(m+1)\,(l+m+3)}$$

$$4\,(l+2)\left(l^2\,m^2-2\,l\,m^3+m^4+2\,l^2\,m+2\,m^3+2\,l^2+8\,l\,m-3\,m^2+8\,l+2\,m+6\right)$$

$$16\left(a^3+\left(\frac{1}{4}\,b^2+b+\frac{7}{4}\right)a^2+\left(\frac{1}{2}\,b^2+3\,b+\frac{5}{2}\right)a+\frac{b^2}{2}+2\,b+\frac{3}{2}\right)(a+b+2) \qquad \textbf{(4.12)}$$

So we drop q(l+3,m-2,2*n) and want to show that what remains is positive;
note that l >= m >= 2:
therefore we set m=a+2 and l=a+b+2 in order to have a,b>=0 independently;
and we see that all terms have non-negative coefficients, which proves the claim.

```
> qe4 := simplify(subs(q(l+3,m-2,2*n)=1,m=a+2,l=a+b+2,qe3));
```

$$qe4 := \left(\left(64\left(a^5+\frac{(b+7)\left(b^2+2\,b+17\right)a^4}{8}+\left(\frac{1}{8}\,b^4+\frac{5}{2}\,b^3+15\,b^2+\frac{647}{8}+\frac{81}{2}\,b\right)a^3+\left(b^4+\frac{115}{8}\,b^3+\frac{557}{8}\,b^2+\frac{1205}{8}\,b+\frac{1571}{8}\right)a^2+\frac{5\left(b^2+\frac{23}{10}\,b+\frac{31}{10}\right)(b+5)^2\,a}{2}+\frac{7\,(b+1)^2\left(b+\frac{29}{7}\right)(b+5)}{4}\right)\left(a+\frac{b}{2}+\frac{9}{2}\right)(a+1)\,q(a+b+4,a+1,2\,n)+8\,(a+2)\,(b+3)^2\left(a^3+\left(\frac{b}{2}+\frac{19}{2}\right)a^2+\left(\frac{b}{2}+\frac{57}{2}\right)a-\frac{b^2}{2}-4\,b+\frac{49}{2}\right)(a+3)\,(b+5)\right)\Big/\left((2\,a+b+5)\,(a+2)^2\,(a+4)\,(b+3)^2\,(a+3)\,(2\,a+b+7)\,(b+5)\,(a+1)\,(2\,a+b+9)\right) \qquad \textbf{(4.13)}$$

```
> 
  subs(l=l+2,m=m-1,qrecUpBound);
```

$$q(l+2,m-1,2\,n)=\frac{(l-m+3)\,q(l+3,m-2,2\,n)}{l-m+5} \qquad \textbf{(4.14)}$$

Finally, we have to treat the case m=1 seperately.
Let us start again at the beginning; note that many terms are zero here

```
> simplify(subs(m=1,l=a+1,qe1)):
  qem11 := simplify(subs(q(a+2,-1,2*n)=0,q(a+4,-1,2*n)=0,q
  (a+2,-1,2*n)=0,q(a+3,-2,2*n)=0,%));
```

$$qem11 := \frac{2\left(a^2+5a+6\right) q(a+2,1,2n)}{3(a+3)(a+1)} - \frac{(a+1)\, q(a+2,3,2n)}{3a+15} - \frac{2\, q(a+3,0,2n)}{3a+9} - \frac{2(a-3)\, q(a+4,1,2n)}{3(a+5)(a+3)} - \frac{4\, q(a+4,3,2n)}{3(a+7)(a+5)} \quad \textbf{(4.15)}$$

```
> qem12 := collect(subs(
      subs(l=a+2,m=1+2,qdown),
      subs(l=a+4,m=1+2,qdown),
  qem11), q,factor);
```

$$qem12 := \frac{2(5a+9)\, q(a+2,1,2n)}{3(a+5)(a+1)} - \frac{2\, q(a+3,0,2n)}{3(a+3)} - \frac{2(a+9)(a-1)\, q(a+4,1,2n)}{3(a+5)(a+3)(a+7)} \quad \textbf{(4.16)}$$

Here the coefficient of q(a+2,1,2n) is positive;
hence we use it to bound term term q(a+3,0,2n);
we use these two bounds bound

```
> subs(l=a+2,m=1,qrecUpBound);
```

$$q(a+2,1,2n) = \frac{(a+1)\, q(a+3,0,2n)}{a+3} \quad \textbf{(4.17)}$$

now the coeff of q(a+3,0,2n) is positive too

```
> qem13 := collect(subs(
      subs(l=a+2,m=1,qrecUpBound),
  qem12), q,factor);
```

$$qem13 := \frac{8(a+1)\, q(a+3,0,2n)}{3(a+5)(a+3)} - \frac{2(a+9)(a-1)\, q(a+4,1,2n)}{3(a+5)(a+3)(a+7)} \quad \textbf{(4.18)}$$

Next, we need to use a different bound, as qrecUpBound would lead to q(a+4,-1,2n)
But we just go back to qrec and use the second instead of the first term on the rhs.
Note again that the = is actually a >=

```
> qrecUpBound2 := q(l, m, 2*n) = (l-m+2)*q(l+1, m+1, 2*n)/(l+
  m+2)
```

$$qrecUpBound2 := q(l,m,2n) = \frac{(l-m+2)\, q(l+1,m+1,2n)}{l+m+2} \quad \textbf{(4.19)}$$

We use this bound (the = is >=)

```
> subs(l=a+3,m=0,qrecUpBound2);
```

$$q(a+3,0,2n) = q(a+4,1,2n) \quad \textbf{(4.20)}$$

And everything is positive :)

```
> qem14 := collect(subs(
      subs(l=a+3,m=0,qrecUpBound2),
  qem13), q,factor);
```

$$qem14 := \frac{2\left(3a^2+24a+37\right) q(a+4,1,2n)}{3(a+5)(a+3)(a+7)} \quad \textbf{(4.21)}$$